
[bookmark: _Toc349662369]Билет 1. 2 Бригады 2 недели на 2 недели, надбавка за Стаж особых работ.
Задача1: Исполнительные листы
Интересен только отчет по удержаниям по исполнительным листам.
Берем отчет «Исполнительные листы», в нем открываем настройку структуры отчета, в верхнем левом углу выбираем «кросс-таблица». В группировках строк: получатель; в группировках колонок: месяц начисления; в колонках: инн и осн. банковский счет получателя, удержано по исполнительным листам.
Ставим флажок «Расширенная настройка», в оформлении указываем «расположение реквизитов» - «отдельно».
Задача 2. 
Доработки:
1) Добавляем документ, отчет, вносим их в интерфейс Полный. 
2) Документ «График выполнения особых работ». Шапка: Организация, ВидРасчета, ПериодРегистрации.
Строки: Сотрудник, День1… День31 (булево). Поставить обязательность для нужных полей.
Обработка проведения по регистру сведений ПлановыеНачисленияРаботниковОрганизаций. Обработка проведения (объединяем периоды, заполняем виды расчета, действия и периоды начала и завершения; цель – чтобы начисления стали плановыми и автоматом подтянулись):
Движения.ПлановыеНачисленияРаботниковОрганизаций.Записывать = Истина;
Движения.ПлановыеНачисленияРаботниковОрганизаций.Очистить();
НачалоПериода = НачалоМесяца(ПериодРегистрации);
Для Каждого ТекСтрокаГрафик Из График Цикл
	МассивПериодов = новый Массив;
	ПредыдущаяДата = Ложь;
	НомерЭлемента = -1;
	Для инд=1 по 31 Цикл
		ТекущаяДата =  ТекСтрокаГрафик["День" + Инд];
		Если ТекущаяДата Тогда
			Если не ПредыдущаяДата Тогда
				// новый период
				ДанныеПериода = новый структура("ДатаНачала,ЧислоДней");
				ДанныеПериода.ДатаНачала = НачалоПериода + (Инд - 1) * 24*60*60;
				ДанныеПериода.ЧислоДней = 1;
				МассивПериодов.Добавить(ДанныеПериода);
				НомерЭлемента = НомерЭлемента + 1;
			Иначе 
				// продолжается период
				МассивПериодов[НомерЭлемента].ЧислоДней =  МассивПериодов[НомерЭлемента].ЧислоДней + 1;
			КонецЕсли;
		КонецЕсли;
		ПредыдущаяДата = ТекущаяДата;
	КонецЦикла;
	Для каждого ДанныеПериода из МассивПериодов Цикл
		Движение = Движения.ПлановыеНачисленияРаботниковОрганизаций.Добавить();
		Движение.Сотрудник = ТекСтрокаГрафик.Сотрудник;
		Движение.Организация = Организация;
		Движение.ВидРасчетаИзмерение = ВидРасчета;
		Движение.ВидРасчета = ВидРасчета;
		
		Движение.Действие = Перечисления.ВидыДействияСНачислением.Начать;
		Движение.ДействиеЗавершения = Перечисления.ВидыДействияСНачислением.Прекратить;
		Движение.Период = ДанныеПериода.ДатаНачала;
		Движение.ПериодЗавершения = Движение.Период + ДанныеПериода.ЧислоДней*24*60*60;
	КонецЦикла;
КонецЦикла;
3) Добавляем предопределенный эл-т спр. ПоказателиСхемМотивации «СтажОсобыхРабот». 
4) Расчет ЗП: в модуле ПроведениеРасчетовПереопределяемый вносим правки:
· Функция ПолучитьСтруктуруНеобходимыхДанных:
в начале:
НеобходимыеДанные.Вставить("СтажОсобыхРабот", Ложь);
в конце в цикле по показателям:
НеобходимыеДанные.СтажОсобыхРабот = НеобходимыеДанные.СтажОсобыхРабот Или Показатель = Справочники.ПоказателиСхемМотивации.СтажОсобыхРабот;
· Функция ПолучитьДанныеДляРасчета:
В след. запросе можно доп-то отфильтровать по начислениям, выбрав только с нашим показателем. А то много лишнего.
СтажОсобыхРаботТекст = 
"ВЫБРАТЬ
|	ВложенныйЗапрос.НомерСтроки КАК НомерСтроки,
|	ВложенныйЗапрос.СтажОР + ЕСТЬNULL(ФизическиеЛицаСтажи.РазмерМесяцев, 0) + ЕСТЬNULL(ФизическиеЛицаСтажи.РазмерДней, 0) / 30.4 КАК СтажОР
|ПОМЕСТИТЬ ВТДанныеОСтажеОсобыхРабот
|ИЗ
|	(ВЫБРАТЬ
|		ОсновныеНачисленияРаботниковОрганизаций.НомерСтроки КАК НомерСтроки,
|		СУММА(ЕСТЬNULL(ДанныеГрафика.ОсновноеЗначениеФактическийПериодДействия, 0) / 30.4) КАК СтажОР,
|		ОсновныеНачисленияРаботниковОрганизаций.ФизЛицо КАК ФизЛицо,
|		ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета.ВидСтажа КАК ВидРасчетаВидСтажа
|	ИЗ
|		РегистрРасчета.ОсновныеНачисленияРаботниковОрганизаций КАК ОсновныеНачисленияРаботниковОрганизаций
|			ЛЕВОЕ СОЕДИНЕНИЕ РегистрРасчета.ОсновныеНачисленияРаботниковОрганизаций.ДанныеГрафика КАК ДанныеГрафика
|			ПО ОсновныеНачисленияРаботниковОрганизаций.Сотрудник = ДанныеГрафика.Сотрудник
|				И ОсновныеНачисленияРаботниковОрганизаций.ПериодДействия > ДанныеГрафика.ПериодДействия
|				И ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета = ДанныеГрафика.ВидРасчета
|	ГДЕ
|		ОсновныеНачисленияРаботниковОрганизаций.Регистратор = &парамРегистратор
|		И ОсновныеНачисленияРаботниковОрганизаций.Авторасчет
|	
|	СГРУППИРОВАТЬ ПО
|		ОсновныеНачисленияРаботниковОрганизаций.НомерСтроки,
|		ОсновныеНачисленияРаботниковОрганизаций.ФизЛицо,
|		ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета.ВидСтажа) КАК ВложенныйЗапрос
|		ЛЕВОЕ СОЕДИНЕНИЕ Справочник.ФизическиеЛица.Стажи КАК ФизическиеЛицаСтажи
|		ПО ВложенныйЗапрос.ФизЛицо = ФизическиеЛицаСтажи.Ссылка
|			И ВложенныйЗапрос.ВидРасчетаВидСтажа = ФизическиеЛицаСтажи.ВидСтажа
|
|ИНДЕКСИРОВАТЬ ПО
|	НомерСтроки";

В ветке по основным начислениям:
Если НеобходимыеДанные.СтажОсобыхРабот Тогда
			Запрос.Текст = СтажОсобыхРаботТекст;
			Запрос.Выполнить();
			
			ТекстЗапроса = ТекстЗапроса + "
			|ЛЕВОЕ СОЕДИНЕНИЕ  ВТДанныеОСтажеОсобыхРабот КАК ДанныеОСтажеОсобыхРабот
			|ПО Основной.НомерСтроки = ДанныеОСтажеОсобыхРабот.НомерСтроки";
КонецЕсли;
· ПолучитьТекстПоказателя
В условие в начале функции добавляем «или НеобходимыеДанные.СтажОсобыхРабот»
Далее в начале сборки текста в нужной ветке:
Показатель1Текст = "ВЫБОР "+
// {экзамен
?(НеобходимыеДанные.СтажОсобыхРабот, "                 
|КОГДА Показатели.Показатель1 = ЗНАЧЕНИЕ(Справочник.ПоказателиСхемМотивации.СтажОсобыхРабот)
|ТОГДА ЕСТЬNULL(ДанныеОСтажеОсобыхРабот.СтажОР,0)", "")+


5) Переходим к вводу данных:
5.1) Добавляем сменные (от слова смена) графики для Бригада1 и Бригада2. Начинать с рабочего дня (в конструкторе), далее напильником первые дни года.
5.2) Ставим флажок ипользования управленческого учета в настройках программы (чтобы полный интерфейс не подменялся)
5.3) Прием на работу сотрудников. Основное начисление – «Оплата по дневному тарифу», не забыть указать размер тарифа. УКАЗАТЬ ПРАВИЛЬНЫЙ ГРАФИК!
5.4) Создаем ОСНОВНОЕ начисление "_Надбавка по особым работам". Делать на основании оплаты по дневному тарифу, чтобы было верное вытеснение (заполненный список вытесняющих). Указать на первой вкладке Вид Стажа (виден, если выбрать начисление "От стажа процентом"). На закладке «время» указать, что это доп. начисление за уже оплаченное время. Заполнить произвольную формулу: 
ОценитьПо(СтажОсобыхРабот,ШакалаОсобыхРабот)*ВремяВДнях*ТарифнаяСтавкаДневная
Где:
СтажОсобыхРабот – наш предопределенный Показатель схем мотивации. Если не виден – в справочнике заполнить его идентификатор.
ШакалаОсобыхРабот – добавляем показатель типа Шкала, числовая. Заполняем значения согласно данным в задании. У первого интервала в «с» указать лучше «-1», т.к. при стаже 0 считает, что в интервал не попали.
5.5) Заполняем пару документов «График выполнения особых работ».
5.6) В справочнике Физические лица указываем стаж работников по нашему виду стажа (указанному в «Вид стажа» начисления). Это для эмуляции «начального» стажа.
5.6) Начисляем зарплату. Если все ок: заполнятся/рассчитаются в том числе надбавки. Причем надбавки будут «умные»: они будут знать, в какие дни мы реально работали по графику (т.к. это доп. начисления за уже оплаченное время), 

6) Отчет: 
ВЫБРАТЬ
	ОсновныеНачисленияРаботниковОрганизаций.ПериодРегистрации КАК Период,
	ОсновныеНачисленияРаботниковОрганизаций.Сотрудник,
	ОсновныеНачисленияРаботниковОрганизаций.Сотрудник.ГрафикРаботы КАК Бригада,
	ВЫБОР
		КОГДА ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета = ЗНАЧЕНИЕ(ПланВидовРасчета.ОсновныеНачисленияОрганизаций.ТарифДневной)
			ТОГДА ОсновныеНачисленияРаботниковОрганизаций.Результат
		ИНАЧЕ 0
	КОНЕЦ КАК НачисленоПоОкладу,
	ВЫБОР
		КОГДА ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета = &ВидРасчета
			ТОГДА ОсновныеНачисленияРаботниковОрганизаций.Результат
		ИНАЧЕ 0
	КОНЕЦ КАК НадбавкаЗаСложность
ИЗ
	РегистрРасчета.ОсновныеНачисленияРаботниковОрганизаций КАК ОсновныеНачисленияРаботниковОрганизаций
ГДЕ
	ОсновныеНачисленияРаботниковОрганизаций.ПериодРегистрации МЕЖДУ &ДатаНач И &ДатаКон
	И (ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета = ЗНАЧЕНИЕ(ПланВидовРасчета.ОсновныеНачисленияОрганизаций.ТарифДневной)
			ИЛИ ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета = &ВидРасчета)


[bookmark: _Toc349662370]Билет 2. Бонусы от суммы продаж процентом
Задача1: Новосельцев, полставки
Для работы Новосельцева на 0.5 ставки по совместительству нужно создать новый график неполного рабочего времени (20 часов, норма времени по другому графику – основному, 40 часов). Назначаем ОКЛАД ПО ЧАСАМ с ставкой из задания. При расчете будет считать верно (отработана половина часов нормы, значит оклад будет делить на 2).
Отчет по плановым начислениям:
УниверсальныйОтчет, выбираем «Плановые начисления сотрудников организаций», в расширенных настройках должно быть:
Группировки: 1) Сотрудник.ФизЛицо; 2) Детальные записи
В полях по физ. лицу поля: № п/п (из системных), Сотрудник.ФизЛицо. Через контекстное меню «Установить заголовок».
В полях по детальным записям: Регистратор; в группе с типом «В отдельной колонке» размещаем: Вид расчета, Размер1.
Снять галку у автополей и там, и там.
В отбор отчета добавить организацию. Выглядит как на картинке, капельку только оформление шапки отличается. Зато данные как на картинке.
Задача2:
Создаем отчет, документ(Организация; ТЧ: Сотрудник, Сумма), включаем их в полный интерфейс.
Создаем оборотный регистр СделкиСотрудниковОрганизаций(Организация; Сотрудник; ресурс Сумма). В документе убираем оперативное проведение, конструируем обработку проведения.
Создаем предопределенный ПоказателиСхемМотивации СуммаСделок.
Расчет ЗП: в модуле ПроведениеРасчетовПереопределяемый вносим правки:
· Функция ПолучитьСтруктуруНеобходимыхДанных:
в начале:
НеобходимыеДанные.Вставить("СуммаСделок", Ложь);
в конце в цикле по показателям:
НеобходимыеДанные.СуммаСделок = НеобходимыеДанные.СуммаСделок Или Показатель = Справочники.ПоказателиСхемМотивации.СуммаСделок;
· Функция ПолучитьДанныеДляРасчета:
	СуммаСделокТекст =
	"ВЫБРАТЬ РАЗЛИЧНЫЕ
	|	ОсновныеНачисленияОрганизацийПоказатели.Ссылка
	|ПОМЕСТИТЬ ВТ_СуммаСделок_ВидыРасчета
	|ИЗ
	|	ПланВидовРасчета.ОсновныеНачисленияОрганизаций.Показатели КАК ОсновныеНачисленияОрганизацийПоказатели
	|ГДЕ
	|	ОсновныеНачисленияОрганизацийПоказатели.Показатель = ЗНАЧЕНИЕ(Справочник.ПоказателиСхемМотивации.СуммаСделок)
	|;
	|
	|////////////////////////////////////////////////////////////////////////////////
	|ВЫБРАТЬ
	|	ОсновныеНачисленияРаботниковОрганизаций.НомерСтроки,
	|	ОсновныеНачисленияРаботниковОрганизаций.Сотрудник КАК Сотрудник,
	|	ОсновныеНачисленияРаботниковОрганизаций.Организация КАК Организация
	|ПОМЕСТИТЬ ВТ_СуммаСделок_Начисления
	|ИЗ
	|	РегистрРасчета.ОсновныеНачисленияРаботниковОрганизаций КАК ОсновныеНачисленияРаботниковОрганизаций
	|		ВНУТРЕННЕЕ СОЕДИНЕНИЕ ВТ_СуммаСделок_ВидыРасчета КАК ВТ_СуммаСделок_ВидыРасчета
	|		ПО ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета = ВТ_СуммаСделок_ВидыРасчета.Ссылка
	|ГДЕ
	|	ОсновныеНачисленияРаботниковОрганизаций.Регистратор = &парамРегистратор
	|	И ОсновныеНачисленияРаботниковОрганизаций.Авторасчет
	|
	|ИНДЕКСИРОВАТЬ ПО
	|	Организация,
	|	Сотрудник
	|;
	|
	|////////////////////////////////////////////////////////////////////////////////
	|ВЫБРАТЬ
	|	ВТ_СуммаСделок_Начисления.НомерСтроки КАК НомерСтроки,
	|	СделкиСотрудниковОрганизацийОбороты.СуммаСделокОборот
	|ПОМЕСТИТЬ ВТСуммаСделок
	|ИЗ
	|	РегистрНакопления.СделкиСотрудниковОрганизаций.Обороты(
	|			&парамПериодРегистрации,
	|			КОНЕЦПЕРИОДА(&парамПериодРегистрации, МЕСЯЦ),
	|			,
	|			(Организация, Сотрудник) В
	|				(ВЫБРАТЬ
	|					ВТ_СуммаСделок_Начисления.Организация,
	|					ВТ_СуммаСделок_Начисления.Сотрудник
	|				ИЗ
	|					ВТ_СуммаСделок_Начисления)) КАК СделкиСотрудниковОрганизацийОбороты
	|		ВНУТРЕННЕЕ СОЕДИНЕНИЕ ВТ_СуммаСделок_Начисления КАК ВТ_СуммаСделок_Начисления
	|		ПО СделкиСотрудниковОрганизацийОбороты.Организация = ВТ_СуммаСделок_Начисления.Организация
	|
	|ИНДЕКСИРОВАТЬ ПО
	|	НомерСтроки"

В ветке по основным начислениям:
Если НеобходимыеДанные.СуммаСделок Тогда
			Запрос.Текст = СуммаСделокТекст;
			Запрос.Выполнить();

			// присоединение к данным о сумме
			ТекстЗапроса = ТекстЗапроса + "
			|ЛЕВОЕ СОЕДИНЕНИЕ ВТСуммаСделок КАК СуммаСделок			
			|ПО Основной.НомерСтроки = СуммаСделок.НомерСтроки";
КонецЕсли;
Дополнительно желательно (но необязательно) заполнить значение показателя в ТЧ документа. Это позволит в отчете использовать только таблицу регистра расчета, без оборотного регистра (если лезть в оборотный регистр – могут снять баллы)
// {Экзамен
Если НеобходимыеДанные.СуммаСделок Тогда
// присоединение к данным о сумме
ТекстЗапроса = ТекстЗапроса + ",
|ЕстьNull(СуммаСделок.СуммаСделокОборот, 0) КАК СуммаСделокОборот,
|ВЫБОР КОГДА СуммаСделок.СуммаСделокОборот is null Тогда Ложь Иначе Истина Конец КАК ЕстьСуммаСделокОборот";
КонецЕсли;
Также в этом случае нужно внести изменения в процедуру РассчитатьЗаписиНабора после «Если СтрокаТабличнойЧасти <> Неопределено Тогда»:
Если ИсходныеДанные.ЕстьСуммаСделокОборот Тогда
	СтрокаТабличнойЧасти.Показатель1 = ИсходныеДанные.СуммаСделокОборот;
КонецЕсли;
· ПолучитьТекстПоказателя
В условие в начале функции добавляем «НеобходимыеДанные.СуммаСделок или»
Далее в начале сборки текста в нужной ветке:
Показатель1Текст = "ВЫБОР "+
// {экзамен
			?(НеобходимыеДанные.СуммаСделок, "                 
			|	КОГДА Показатели.Показатель1 = ЗНАЧЕНИЕ(Справочник.ПоказателиСхемМотивации.СуммаСделок)
			|		ТОГДА ЕстьNull(СуммаСделок.СуммаСделокОборот, 0)", "") +

Отчет (для случая хранения суммы сделок в поле Показатель1; если не хранить – добавляем соединение с регистром оборотным (периодичность Месяц, соединяем по периоду)):
ВЫБРАТЬ
	ОсновныеНачисленияРаботниковОрганизаций.Сотрудник,
	ВЫБОР
		КОГДА ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета <> ЗНАЧЕНИЕ(ПланВидовРасчета.ОсновныеНачисленияОрганизаций.ОкладПоДням)
			ТОГДА ОсновныеНачисленияРаботниковОрганизаций.Показатель1
		ИНАЧЕ 0
	КОНЕЦ КАК СуммаСделок,
	ВЫБОР
		КОГДА ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета = ЗНАЧЕНИЕ(ПланВидовРасчета.ОсновныеНачисленияОрганизаций.ОкладПоДням)
			ТОГДА ОсновныеНачисленияРаботниковОрганизаций.Результат
		ИНАЧЕ 0
	КОНЕЦ КАК НачисленоПоОкладу,
	ВЫБОР
		КОГДА ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета <> ЗНАЧЕНИЕ(ПланВидовРасчета.ОсновныеНачисленияОрганизаций.ОкладПоДням)
			ТОГДА ОсновныеНачисленияРаботниковОрганизаций.Результат
		ИНАЧЕ 0
	КОНЕЦ КАК СуммаБонуса,
	ОсновныеНачисленияРаботниковОрганизаций.ПериодРегистрации
ИЗ
	РегистрРасчета.ОсновныеНачисленияРаботниковОрганизаций КАК ОсновныеНачисленияРаботниковОрганизаций
ГДЕ
	ОсновныеНачисленияРаботниковОрганизаций.Активность
	И ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета В (ЗНАЧЕНИЕ(ПланВидовРасчета.ОсновныеНачисленияОрганизаций.ОкладПоДням), &Бонус)
В вычисляемых полях считаем процент бонуса:
ВЫбор когда СуммаСделок<1000000 Тогда 5 Когда СуммаСделок<3000000 тогда 10 иначе 15 конец
Работник идет в строки, Период – в колонки. Для полей не забыть объединить в группу 3 поля, относящиеся к бонусу, и назвать группу «Бонус».

Ввод данных: в новом начислении вводим формулу:
СуммаСделок*?(СуммаСделок<1000000,0.05,?(СуммаСделок<3000000,0.10,0.15))
Либо формулу оценки по шкале.



[bookmark: _Toc349662371]Билет 3. Материальная выгода за обеды в столовой. 
СМОТРЕТЬ БИЛЕТ 6 – они похожы, но задача 2 решена правильнее.
Задача1: ПолисыОМС
Реализуем через доп. свойства сотрудников. Причина: в специализированном регистре медицинского страхования нельзя завести полис ОМС с датой начала, раньше чем дата приема сотрудника; что как-то может не уложиться в исходную задачу. Если все же  через регистр медицинского страхования, то вместо переделки стандартного отчета лучше взять универсальный отчет по данным регистра сведений (стандартный оказалось сложно переделать, чтобы уместить оба отчета на одном листе).
Отчет: за основу берем «Списки сотрудников организации».  Делаем в нем пользовательское поле «ЕстьПолис»:
Выбор
	Когда ЕстьNull([Сотрудник. Дополнительное свойство: Номер полиса ОМС], "") = ""
		Тогда Ложь
	Иначе Истина
Конец
Далее два простых отчета (две группировки по детальным записям; поля группируем как на картинке).
Задача2.
Добавляем:
Справочник ВидыОбедов
РС (период месяц) «ЛьготныеЦеныОбедов»: Обед; ЛьготнаяЦена
В РР ДополнительныеНачисленияРаботниковОрганизаций добавить реквизит «ВидОбеда».
Документ: ОбедыВСтоловой. Оперативное проведение запретить.
Реквизиты: Организация, ВидРасчета (ДополнительныеНачисленияОрганизаций)
Таб. части: а) РыночныеСтоимостиОбедов (обед/Цена),  ОтпущенныеОбеды (Сотрудник, ДеньДекады1…ДеньДекады11 – ссылка на вид обеда)
Обработка проведения:
// вычислим суммарное число обедов по сотрудникам по видам обедов
ТаблицаЧислаОбедовСотрудников = новый ТаблицаЗначений;
ТаблицаЧислаОбедовСотрудников.Колонки.Добавить("Сотрудник",новый ОписаниеТипов("СправочникСсылка.СотрудникиОрганизаций"));
ТаблицаЧислаОбедовСотрудников.Колонки.Добавить("ВидОбеда",новый ОписаниеТипов("СправочникСсылка.ВидыОбедов"));
ТаблицаЧислаОбедовСотрудников.Колонки.Добавить("Количество",новый ОписаниеТипов("Число"));
Для каждого СтрокаОбеда Из ОтпущенныеОбеды Цикл
	КоличествоОбедовСотрудника = новый Соответствие;
	Для Сч=1 По 11 Цикл
		ВидОбеда = СтрокаОбеда["ДеньДекады"+Строка(Сч)];
		Если ЗначениеЗаполнено(ВидОбеда) Тогда
			ТекКолво = КоличествоОбедовСотрудника[ВидОбеда];
			Если ТекКолво=неопределено Тогда
				ТекКолво = 0;
			КонецЕсли;
			КоличествоОбедовСотрудника[ВидОбеда] = ТекКолво+1;
		КонецЕсли;
	КонецЦикла;
	Для каждого ДанныеПоОбеду из КоличествоОбедовСотрудника Цикл
		НовСтр = ТаблицаЧислаОбедовСотрудников.Добавить();
		НовСтр.Сотрудник = СтрокаОбеда.Сотрудник;
		НовСтр.ВидОбеда = ДанныеПоОбеду.Ключ;
		НовСтр.Количество = ДанныеПоОбеду.Значение;
	КонецЦикла;
КонецЦикла;
ЗапросМатериальнаяВыгода = новый Запрос();
ЗапросМатериальнаяВыгода.Текст = "ВЫБРАТЬ
|	ОбедыСотрудников.Сотрудник,
|	ОбедыСотрудников.ВидОбеда КАК ВидОбеда,
|	ОбедыСотрудников.Количество
|ПОМЕСТИТЬ ВТОбедыСотрудников
|ИЗ
|	&ОбедыСотрудников КАК ОбедыСотрудников
|
|ИНДЕКСИРОВАТЬ ПО
|	ВидОбеда
|;
|
|////////////////////////////////////////////////////////////////////////////////
|ВЫБРАТЬ
|	ВТОбедыСотрудников.Сотрудник,
|	ВТОбедыСотрудников.Сотрудник.ФизЛицо как ФизЛицо,
|	ВТОбедыСотрудников.ВидОбеда,
|	ВЫБОР
|		КОГДА РаботникиОрганизации.ПериодЗавершения <= &ДатаДок
|				И РаботникиОрганизации.ПериодЗавершения <> ДАТАВРЕМЯ(1, 1, 1, 0, 0, 0)
|			ТОГДА РаботникиОрганизации.ПодразделениеОрганизацииЗавершения
|		ИНАЧЕ РаботникиОрганизации.ПодразделениеОрганизации
|	КОНЕЦ КАК ПодразделениеОрганизации,
|	СУММА(ВТОбедыСотрудников.Количество) КАК Количество,
|	МАКСИМУМ(ЛьготныеЦеныОбедовСрезПоследних.ЛьготнаяЦена) КАК ЛьготнаяЦена,
|	МАКСИМУМ(ОбедыВСтоловойРыночныеСтоимостиОбедов.Цена) КАК ЦенаРыночная
|ИЗ
|	ВТОбедыСотрудников КАК ВТОбедыСотрудников
|		ЛЕВОЕ СОЕДИНЕНИЕ Документ.ОбедыВСтоловой.РыночныеСтоимостиОбедов КАК ОбедыВСтоловойРыночныеСтоимостиОбедов
|		ПО ВТОбедыСотрудников.ВидОбеда = ОбедыВСтоловойРыночныеСтоимостиОбедов.Обед
|			И (ОбедыВСтоловойРыночныеСтоимостиОбедов.Ссылка = &Ссылка)
|		ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЛьготныеЦеныОбедов.СрезПоследних(
|				&МоментВремени,
|				Обед В
|					(ВЫБРАТЬ РАЗЛИЧНЫЕ
|						ВТОбедыСотрудников.ВидОбеда
|					ИЗ
|						ВТОбедыСотрудников КАК ВТОбедыСотрудников)) КАК ЛьготныеЦеныОбедовСрезПоследних
|		ПО (ОбедыВСтоловойРыночныеСтоимостиОбедов.Обед = ЛьготныеЦеныОбедовСрезПоследних.Обед)
|		ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.РаботникиОрганизаций.СрезПоследних(
|				&МоментВремени,
|				Сотрудник В
|					(ВЫБРАТЬ РАЗЛИЧНЫЕ
|						ВТОбедыСотрудников.Сотрудник
|					ИЗ
|						ВТОбедыСотрудников КАК ВТОбедыСотрудников)) КАК РаботникиОрганизации
|		ПО (РаботникиОрганизации.Сотрудник = ВТОбедыСотрудников.Сотрудник)
|
|СГРУППИРОВАТЬ ПО
|	ВТОбедыСотрудников.Сотрудник,
|	РаботникиОрганизации.ПодразделениеОрганизации,
|	РаботникиОрганизации.ПериодЗавершения,
|	РаботникиОрганизации.ПодразделениеОрганизацииЗавершения,
|	ВТОбедыСотрудников.ВидОбеда";

ЗапросМатериальнаяВыгода.УстановитьПараметр("Ссылка",Ссылка);
ЗапросМатериальнаяВыгода.УстановитьПараметр("МоментВремени",МоментВремени());
ЗапросМатериальнаяВыгода.УстановитьПараметр("ДатаДок",Дата);
ЗапросМатериальнаяВыгода.УстановитьПараметр("ОбедыСотрудников",ТаблицаЧислаОбедовСотрудников);
Выборка = ЗапросМатериальнаяВыгода.Выполнить().Выбрать();
Движения.ДополнительныеНачисленияРаботниковОрганизаций.Записывать = Истина;
Движения.ДополнительныеНачисленияРаботниковОрганизаций.Очистить();
Движения.НДФЛСведенияОДоходах.Записывать = Истина;
Движения.НДФЛСведенияОДоходах.Очистить();
КодДоходаНДФЛ = ВидРасчета.КодДоходаНДФЛ;
Пока Выборка.Следующий() Цикл
	Если Выборка.ЦенаРыночная=null Тогда
		ОбщегоНазначения.СообщитьОбОшибке("Не указана рыночная цена обеда "+Выборка.ВидОбеда,Отказ);
	КонецЕсли;
	Если Выборка.ЛьготнаяЦена=null Тогда
		ОбщегоНазначения.СообщитьОбОшибке("Не указана льготная цена обеда "+Выборка.ВидОбеда,Отказ);
	КонецЕсли;
	Если не Отказ Тогда
		МатериальнаяВыгода = Выборка.Количество*(Выборка.ЦенаРыночная-Выборка.ЛьготнаяЦена);
		Если МатериальнаяВыгода>0 Тогда
			// регистр ДополнительныеНачисленияРаботниковОрганизаций
			Движение = Движения.ДополнительныеНачисленияРаботниковОрганизаций.Добавить();
			Движение.ВидРасчета = ВидРасчета;
			Движение.ПериодРегистрации = НачалоМесяца(Дата);
			Движение.Сотрудник = Выборка.Сотрудник;
			Движение.ФизЛицо = Выборка.ФизЛицо;
			Движение.Организация = Организация;
			Движение.Результат = МатериальнаяВыгода;
			//Движение.Показатель1 = МатериальнаяВыгода;
			Движение.Показатель1 = Выборка.Количество;
			Движение.ВидОбеда = Выборка.ВидОбеда;
			Движение.ПодразделениеОрганизации = Выборка.ПодразделениеОрганизации;
			Движение.ОбособленноеПодразделение = Организация;
			Движение.ВидУчетаВремени = Перечисления.ВидыУчетаВремени.ПоДням;
			
			
			// НДФЛСведенияОДоходах
			Движение = Движения.НДФЛСведенияОДоходах.Добавить();
			Движение.Период = КонецМесяца(Дата);
			Движение.Организация = Организация;
			Движение.ФизЛицо = Выборка.ФизЛицо;
			Движение.КодДохода = КодДоходаНДФЛ;
			Движение.ПериодРегистрации = КонецМесяца(Дата);
			Движение.СуммаДохода = МатериальнаяВыгода;
			Движение.ОбособленноеПодразделение = Организация;
			Движение.ПодразделениеОрганизации = Выборка.ПодразделениеОрганизации;
			Движение.ВидРасчета = ВидРасчета;
			// Часть остальных реквизитов лучше заполнить как в  СформироватьДоходыПоКодамНДФЛ() из начисления ЗП
		КонецЕсли;
	КонецЕсли;
КонецЦикла;


Как вариант, можно начисление сделать предопределенным. Тогда:
обм=ПланыВидовРасчета.ДополнительныеНачисленияОрганизаций.МатВыгодаПоОбедам.ПолучитьОбъект(); 
  Обм.СпособРасчета=Перечисления.СпособыРасчетаОплатыТруда.ФиксированнойСуммой; 
  Обм.ЯвляетсяДоходомВНатуральнойФорме=Истина; 
  обм.КатегорияРасчета =Справочники.КатегорииРасчетов.Первичное; 
  Обм.ВидПремии=Перечисления.ВидыПремии.НеПремия; 
  обм.ВидНачисленияПоСт255НК=Перечисления.ВидыНачисленийОплатыТрудаПоСт255НК.пп4ст255; 
  Обм.КодДоходаНДФЛ=Справочники.ДоходыНДФЛ.2630; 
  Обм.КодДоходаСтраховыеВзносы=Справочники.ДоходыПоСтраховымВзносам.ОблагаетсяЦеликом; 
  Обм.Записать();

Отчет:
ВЫБРАТЬ
	ДополнительныеНачисленияРаботниковОрганизаций.ПериодРегистрации КАК Период,
	ДополнительныеНачисленияРаботниковОрганизаций.Сотрудник,
	ДополнительныеНачисленияРаботниковОрганизаций.Показатель1 КАК Количество,
	ДополнительныеНачисленияРаботниковОрганизаций.Результат,
	ДополнительныеНачисленияРаботниковОрганизаций.ВидОбеда
ИЗ
	РегистрРасчета.ДополнительныеНачисленияРаботниковОрганизаций КАК ДополнительныеНачисленияРаботниковОрганизаций
ГДЕ
	ДополнительныеНачисленияРаботниковОрганизаций.ПериодРегистрации МЕЖДУ &ПериодНачало И &ПериодКонец
	И ДополнительныеНачисленияРаботниковОрганизаций.ВидРасчета = &ВидРасчета
	И ДополнительныеНачисленияРаботниковОрганизаций.Организация = &Организация

Ввод данных: Вводим новое дополнительное начисление: _ВыгодаПоОбедам, Фиксированной суммой, Является доходом в натуральной форме, НДФЛ – наверно код 2530 (или 2630?), Взносы и ЕСН –облагается вроде бы.
Вводим документ Обедов. 
Начисление НДФЛ производим стандартным документом начисления ЗП (НЗСО – либо полный расчет, либо расчет НДФЛ).


[bookmark: _Toc349662372]Билет 4. Старатели, надбавка от стажа в организации по графику особых работ
Задача1: Договоры подряда
Отчет: «Списки договорников организаций». Настраивается легко и просто.
Задача2. 
Делаем документ, пишем в ПлановыеНачисления на каждый день отдельными записями (иначе не вычислить точный стаж на конкретную дату работ). Обработка проведения:
ДатаНачалаЗаписи = НачалоДня(ПервыйДеньНеделиРабот);
Движения.ПлановыеНачисленияРаботниковОрганизаций.Записывать = Истина;
Движения.ПлановыеНачисленияРаботниковОрганизаций.Очистить();
МинутВДне=24*60*60;
Для Каждого ТекСтрокаГрафикРабот Из ГрафикРабот Цикл
	Для Счетчик=1 по 7 Цикл
		Если ТекСтрокаГрафикРабот["День"+Счетчик] Тогда
			Движение = Движения.ПлановыеНачисленияРаботниковОрганизаций.Добавить();
			Движение.Сотрудник = ТекСтрокаГрафикРабот.Сотрудник;
			Движение.Организация = Организация;
			Движение.ВидРасчетаИзмерение = ВидНачисления;
			Движение.ВидРасчета = ВидНачисления;
			
			Движение.Действие = Перечисления.ВидыДействияСНачислением.Начать;
			Движение.ДействиеЗавершения = Перечисления.ВидыДействияСНачислением.Прекратить;
			Движение.Период = ДатаНачалаЗаписи+(Счетчик-1)*МинутВДне;
			Движение.ПериодЗавершения = Движение.Период+МинутВДне;
		КонецЕсли;
	КонецЦикла;
КонецЦикла;
Как ни странно, без создания нового предопределенного показателя мотивации не обойтись. Причина: предопределенный показатель Стаж считается крайне неподходящим образом: фактически это число целых месяцев, причем относительно даты конца месяца периода начисления. А нужно с точностью на день начисления для начала периода начисления.
Итог: добавляем предопределенный показатель СтажТочный.
В модуле ПроведениеРасчетовПереопределяемый вносим правки:
· Функция ПолучитьСтруктуруНеобходимыхДанных:
в начале:
НеобходимыеДанные.Вставить("СтажТочный", Ложь);
в конце в цикле по показателям:
НеобходимыеДанные.СтажТочный = НеобходимыеДанные.СтажТочный Или Показатель = Справочники.ПоказателиСхемМотивации.СтажТочный;
меняем строку НеобходимыеДанные.Стаж.... на:
НеобходимыеДанные.Стаж = НеобходимыеДанные.Стаж Или Показатель = Справочники.ПоказателиСхемМотивации.Стаж Или Показатель = Справочники.ПоказателиСхемМотивации.СтажТочный;
· ПолучитьТекстПоказателя добавляем:
			?(НеобходимыеДанные.СтажТочный, "                 
			|	КОГДА Показатели.Показатель1 = ЗНАЧЕНИЕ(Справочник.ПоказателиСхемМотивации.СтажТочный)
			|		ТОГДА РАЗНОСТЬДАТ(НачалоПериода(ДанныеОСтаже.Период,Месяц), НачалоПериода(Основной.ПериодДействияНачало,Месяц), МЕСЯЦ)+Выбор когда День(Основной.ПериодДействияНачало)<День(ДанныеОСтаже.Период) Тогда -1 Иначе 0 конец", "") + 
В режиме предприятия добавляем новый Основной вид расчета, зависимый первого уровня, база по окладу по дням, с формулой:
ОценитьПо(СтажТочный,ШкалаНадбавокОтСтажа)*РасчетнаяБаза*(ВремяВДнях/ВремяВДнях)
В шкале (создаем показатель) нужно учесть, что при формировании условия «с» учитывается по «>», а «по» «по» считается по условию «<=».
Для чего  ВремяВДнях/ВремяВДнях: чтобы заполнилось «Оплачено дней/часов» (считать же правильно по базе).
Отчет:
ВЫБРАТЬ
	ОсновныеНачисленияРаботниковОрганизаций.ПериодРегистрации КАК Период,
	ОсновныеНачисленияРаботниковОрганизаций.Сотрудник,
	ОсновныеНачисленияРаботниковОрганизаций.Сотрудник.ГрафикРаботы КАК Бригада,
	ОсновныеНачисленияРаботниковОрганизаций.Результат КАК Надбавка,
	ОсновныеНачисленияРаботниковОрганизаций.ОплаченоДнейЧасов как ОтработаноДней
ИЗ
	РегистрРасчета.ОсновныеНачисленияРаботниковОрганизаций КАК ОсновныеНачисленияРаботниковОрганизаций
ГДЕ
	ОсновныеНачисленияРаботниковОрганизаций.ПериодРегистрации МЕЖДУ &ДатаНач И &ДатаКон
	И ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета = &ВидРасчета

Задача1: Отчет о возврасте сотрудников
Открываем стандартный отчет Списки сотрудников организации
За основу берем вариант «Список сотрудников организации»
В пользовательских полях редактируем поле «Возраст интервалами» согласно заданию.
Для вывода количества используем существующий показатель «Количество».
Дале рисуем таблицу, в столбцах из раздела «Данные о рабочем месте» берем Должность.Категория статистического учета. В строках - «Возраст интервалами». В выводимые поля добавляем Количество. Убираем итоги по горизонтали.
Задача 2. За основу решение WhiteCroW с чудесного всем известного форума. Но: в нем на мой взгляд не хватает реализации перерасчета. Ведь продажи, ни для кого не секрет, будут порой вводить задним числом, что нужно учесть.
1. Создать регистр накопления Продажи (тип регистра Обороты). В нем измерения Организация, Сотрудник, ресурсы - Количество, СуммаПродаж. 
2. Создать документ Регистрация продаж, где реквизиты Организация, в табличной части реквизиты Сотрудник, СуммаПродаж. Этот документ делает проводки только по регистру Продажи(можно заполнить конструктором, количество = 1), а также заполяет перерасчет (на случай, если продажи затронут уже рассчитанное). Примерно так:
// регистр ПродажиАгентов 
Движения.ПродажиАгентов.Записывать = Истина;
Движения.ПродажиАгентов.Очистить();
Для Каждого ТекСтрокаПродажиСотрудников Из ПродажиСотрудников Цикл
	Движение = Движения.ПродажиАгентов.Добавить();
	Движение.Период = Дата;
	Движение.Организация = Организация;
	Движение.Сотрудник = ТекСтрокаПродажиСотрудников.Сотрудник;
	Движение.Количество = 1;
	Движение.Сумма = ТекСтрокаПродажиСотрудников.СуммаПродаж;
КонецЦикла;
// заполним перерасчет
Движения.Записать();
//Запрос по записям регистра расчета
Запрос = Новый Запрос("ВЫБРАТЬ РАЗЛИЧНЫЕ
|	ПродажиАгентов.Сотрудник,
|	ПродажиАгентов.Организация,
|	ДополнительныеНачисленияРаботниковОрганизаций.ФизЛицо КАК ФизЛицо,
|	ДополнительныеНачисленияРаботниковОрганизаций.Регистратор КАК Регистратор
|ИЗ
|	РегистрНакопления.ПродажиАгентов КАК ПродажиАгентов
|		ЛЕВОЕ СОЕДИНЕНИЕ РегистрРасчета.ДополнительныеНачисленияРаботниковОрганизаций КАК ДополнительныеНачисленияРаботниковОрганизаций
|		ПО ПродажиАгентов.Сотрудник = ДополнительныеНачисленияРаботниковОрганизаций.Сотрудник
|			И ПродажиАгентов.Организация = ДополнительныеНачисленияРаботниковОрганизаций.Организация
|			И (ПродажиАгентов.Период МЕЖДУ ДополнительныеНачисленияРаботниковОрганизаций.ПериодРегистрации И КонецПериода(ДополнительныеНачисленияРаботниковОрганизаций.ПериодРегистрации,Месяц))
|ГДЕ
|	ДополнительныеНачисленияРаботниковОрганизаций.ВидРасчета = Значение(ПланВидовРасчета.ДополнительныеНачисленияОрганизаций.ВознагражденияАгентов)
|	И ПродажиАгентов.Регистратор = &Регистратор
|ИТОГИ ПО
|	Регистратор");
Запрос.УстановитьПараметр("Регистратор", Ссылка);
ВыборкаДокументов = Запрос.Выполнить().Выбрать(ОбходРезультатаЗапроса.ПоГруппировкам);
Перерасчет = РегистрыРасчета.ДополнительныеНачисленияРаботниковОрганизаций.Перерасчеты.ПерерасчетДополнительныхНачислений;
Пока ВыборкаДокументов.Следующий() Цикл
	НаборЗаписей = Перерасчет.СоздатьНаборЗаписей();
	НаборЗаписей.Отбор.ОбъектПерерасчета.Значение = ВыборкаДокументов.Регистратор;
	
	Выборка = ВыборкаДокументов.Выбрать();
	Пока Выборка.Следующий() Цикл
		Запись = НаборЗаписей.Добавить();
		Запись.Сотрудник = Выборка.Сотрудник;
		Запись.Организация = Выборка.Организация;
		Запись.ФизЛицо = Выборка.ФизЛицо;
	КонецЦикла;	
	//Запись в таблицу без замещения
	НаборЗаписей.Записать(Ложь);
КонецЦикла;
3. В перечисление СпособыРасчетаОплатыТруда добавить значение ПроцентомОтПродаж. 
4. В Планы видов расчета ДополнительныеНачисленияОрганизаций добавить предопределенный элемент ВознагражденияАгентов. Его закрытые значения заполнить внешней обработкой, отражение в бух и налоговом учете можно заполнить руками:
	ВР = ПланыВидовРасчета.ДополнительныеНачисленияОрганизаций. ВознагражденияАгентов.ПолучитьОбъект();
	ВР.СпособРасчета = Перечисления.СпособыРасчетаОплатыТруда.ПроцентомОтПродаж;
	ВР.КатегорияРасчета = Справочники.КатегорииРасчетов.Первичное;
	ВР.Записать();
5.В документ ДоговорНаВыполнениеРаботСФизЛицом добавляем реквизит Процент, его же выводим на форму. В процедуре ДобавитьСтрокуВПлановыеНачисления модуля документа добавляем:
Движение.Показатель2 = Процент
В форме документа в процедуре ВидРасчетаНачалоВыбора строки: 
ФормаВыбораВидаРасчета.Отбор.СпособРасчета.ВидСравнения = ВидСравнения.Равно; 
  ФормаВыбораВидаРасчета.Отбор.СпособРасчета.Значение = Перечисления.СпособыРасчетаОплатыТруда.ПоДоговоруФиксированнойСуммой;
Заменяем на: 
ФормаВыбораВидаРасчета.Отбор.СпособРасчета.ВидСравнения = ВидСравнения.ВСписке; 
  Список = Новый СписокЗначений; 
  Список.Добавить(Перечисления.СпособыРасчетаОплатыТруда.ПоДоговоруФиксированнойСуммой); 
  Список.Добавить(Перечисления.СпособыРасчетаОплатыТруда.ПроцентомОтПродаж); 
  ФормаВыбораВидаРасчета.Отбор.СпособРасчета.Значение = Список;
Добавляем сам процент на форму.
6. В модуле документа НачислениеЗарплатыРаботникамОрганизации ищем процедуру ВыполнитьАвтозаполнение, в ней ищем строки: 
// работающие по договорам подряда 
Ниже этих строк запрос, который надо править, а именно в ЗапросПакета5 добавим получение продаж:
|ЛЕВОЕ СОЕДИНЕНИЕ РегистрНакопления. ПродажиАгентов.Обороты(&парамНачало, КонецПериода(&парамНачало,Месяц), 
|	, Организация = &парамОрганизация) КАК Продажи ПО СписокНачислений.Сотрудник = Продажи.Сотрудник
и в выбранных полях изменить функции Размер и Результат на следующую формулу (добавляем после слова ВЫБОР): 
|когда СписокНачислений.ВидРасчета = ЗНАЧЕНИЕ(ПланВидовРасчета.ДополнительныеНачисленияОрганизаций.ВознагражденияАгентов) тогда СписокНачислений.Показатель2 * ЕСТЬNULL(Продажи.СуммаОборот, 0) / 100 
Далее протянуть получение Показатель2 выше по запросам и временным таблицам.
7. В модуле документа НачислениеЗарплатыРаботникамОрганизации правим ошибки конфы в процедуре ЗаполнитьПоПерерассчитываемомуДокументу:
В условии «Если Сотрудники = "ПоДаннымПерерасчета" Тогда» во втором запросе добавить получение сотрудник/физлицо из ДоговорыНаВыполнениеРабот (из основых и дополнительных 1С-цы сделали, а про договорняков забыли – добавляем по образце дополнитльных). 
В результате: список сотрудников сформируется верно, и при перерасчете, если использовать «создать документ перерасчета в текущем периоде», то в новый документ успешно будут попадать старые записи как сторно, и будут считаться новые записи по новым продажам. Эврика!

Перерасчет текущего периода: реализовать в теории можно, но тут один косяк: 1С-цы почему-то для дополнительных начислений (в отличие от основных) при перерасчете не возвращают результат расчета в таб. часть. А потом радостно из таб. части берут данные для взаиморасчетов. Красота да и только. И править там все настолько долго и нужно, что пошло оно…

8. Заполнить обработкой реквизиты начисления. Заполнить вручную остальные поля (код дохода 2010)
Принять двух сотрудников по договорам подряда, создать им по документу ДоговорНаВыполнениеРаботСФизЛицами, где вид начисления Вознаграждения работников, указать срок действия договора, выплачивать ежемесячно, заполнить процент. Внести по сотрудникам продажи. Заполнить документ Начисление ЗП. 

9. Отчет формируется по регистру накопления Продажи + РегРасчета.ДополнительныеНачисленияРаботниковОрганизаций.
ВЫБРАТЬ
	ДополнительныеНачисленияРаботниковОрганизаций.Сотрудник,
	ДополнительныеНачисленияРаботниковОрганизаций.ПериодРегистрации,
	ДополнительныеНачисленияРаботниковОрганизаций.Результат,
	ПродажиОбороты.КоличествоОборот,
	ПродажиОбороты.СуммаОборот как СуммаПродажОборот
ИЗ
	РегистрРасчета.ДополнительныеНачисленияРаботниковОрганизаций КАК ДополнительныеНачисленияРаботниковОрганизаций
		ЛЕВОЕ СОЕДИНЕНИЕ РегистрНакопления.ПродажиАгентов.Обороты(, , Месяц, ) КАК ПродажиОбороты
		ПО ДополнительныеНачисленияРаботниковОрганизаций.ПериодРегистрации = ПродажиОбороты.Период
			И ДополнительныеНачисленияРаботниковОрганизаций.Сотрудник = ПродажиОбороты.Сотрудник
			И ДополнительныеНачисленияРаботниковОрганизаций.Организация = ПродажиОбороты.Организация
ГДЕ
	ДополнительныеНачисленияРаботниковОрганизаций.ВидРасчета = &ВидРасчета

10. На всякий случай описание реализации перерасчета текущим периодом (но как писал выше, взаиморасчеты неверные получатся; делать на экзамене на мой взгляд не надо):
Реквизит Авторасчет в ТЧ договоров в документе начисления ЗП.
В модуле документа расчета ЗП в процедуре СформироватьЗапросПоДоговорам  меняем «|	ЛОЖЬ КАК Авторасчет,» на 
 «|	Авторасчет КАК Авторасчет, 
|	СтрокиНачисления.Размер КАК Показатель1,»,
ВыполнитьАвтозаполнение:
|	ВЫБОР КОГДА СписокНачислений.ВидРасчета.СпособРасчета = Значение(Перечисление.СпособыРасчетаОплатыТруда.ПроцентомОтПродаж) Тогда Истина Иначе ЛОЖЬ КОНЕЦ КАК Авторасчет,

Далее в ПроведениеРасчетовПереопределяемый вносим изменения:
Б) ПолучитьСтруктуруНеобходимыхДанных – новое поле необходимых данных «ПродажиАгентов» в зависимости от способа расчета (ПроцентомОтПродаж)
В) ПолучитьДанныеДляРасчета: добавляем получение нужных данных при условии ПродажиАгентов=Истина
ПродажиАгентовТекст = "ВЫБРАТЬ
|	ДополнительныеНачисленияРаботниковОрганизаций.НомерСтроки,
|	ДополнительныеНачисленияРаботниковОрганизаций.Сотрудник КАК Сотрудник,
|	ДополнительныеНачисленияРаботниковОрганизаций.Организация КАК Организация
|ПОМЕСТИТЬ ВТ_СуммаСделок_Начисления
|ИЗ
|	РегистрРасчета.ДополнительныеНачисленияРаботниковОрганизаций КАК ДополнительныеНачисленияРаботниковОрганизаций
|ГДЕ
|	ДополнительныеНачисленияРаботниковОрганизаций.Регистратор = &парамРегистратор
|	И ДополнительныеНачисленияРаботниковОрганизаций.Авторасчет
|	И ДополнительныеНачисленияРаботниковОрганизаций.ВидРасчета.СпособРасчета = ЗНАЧЕНИЕ(Перечисление.СпособыРасчетаОплатыТруда.ПроцентомОтПродаж)
|
|ИНДЕКСИРОВАТЬ ПО
|	Организация,
|	Сотрудник
|;
|////////////////////////////////////////////////////////////////////////////////
|ВЫБРАТЬ
|	ВТ_СуммаСделок_Начисления.НомерСтроки КАК НомерСтроки,
|	СделкиСотрудниковОрганизацийОбороты.СуммаОборот КАК СуммаСделокОборот
|ПОМЕСТИТЬ ВТСуммаСделок
|ИЗ
|	РегистрНакопления.ПродажиАгентов.Обороты(
|			&парамПериодРегистрации,
|			КОНЕЦПЕРИОДА(&парамПериодРегистрации, МЕСЯЦ),
|			,
|			(Организация, Сотрудник) В
|				(ВЫБРАТЬ
|					ВТ_СуммаСделок_Начисления.Организация,
|					ВТ_СуммаСделок_Начисления.Сотрудник
|				ИЗ
|					ВТ_СуммаСделок_Начисления)) КАК СделкиСотрудниковОрганизацийОбороты
|		ВНУТРЕННЕЕ СОЕДИНЕНИЕ ВТ_СуммаСделок_Начисления КАК ВТ_СуммаСделок_Начисления
|		ПО СделкиСотрудниковОрганизацийОбороты.Организация = ВТ_СуммаСделок_Начисления.Организация
|		И СделкиСотрудниковОрганизацийОбороты.Сотрудник = ВТ_СуммаСделок_Начисления.Сотрудник
|
|ИНДЕКСИРОВАТЬ ПО
|	НомерСтроки";
В ветке по доп. начислениям:
Если НеобходимыеДанные.ПродажиАгентов Тогда
ТекстЗапроса = ТекстЗапроса + ",
|ЕстьNull(ПродажиАгентов.СуммаСделокОборот, 0) КАК СуммаСделокОборот,
|ЕстьNull(Основной.Показатель1, 0) КАК ПроцентОтПродаж";
КонецЕсли;
И сама таблица:
Если НеобходимыеДанные.ПродажиАгентов Тогда
	Запрос.Текст = ПродажиАгентовТекст;
	Запрос.Выполнить();	
// присоединение к данным о продажах
	ТекстЗапроса = ТекстЗапроса + "
	|ЛЕВОЕ СОЕДИНЕНИЕ ВТСуммаСделок КАК ПродажиАгентов			
	|ПО Основной.НомерСтроки = ПродажиАгентов.НомерСтроки";
КонецЕсли;
Г) РассчитатьЗаписьРегистраРасчета: добавляем:
ИначеЕсли СпособРасчета = Перечисления.СпособыРасчетаОплатыТруда.ПроцентомОтПродаж Тогда
Результат = ИсходныеДанные.СуммаСделокОборот*ИсходныеДанные.ПроцентОтПродаж/100
Если совсем делать нефиг: разобраться возвратом данных перерасчета в ТЧ документа. Там реализовано только добавление строк в доп. начисления (и даже правки доп. начислений нет). Но на мой взгляд правка ошибок 1С-цев в задачу на экзамене не входит.

Задача1: Отчет о возрасте сотрудников гистограммой
Открываем стандартный отчет Списки сотрудников организации
За основу берем вариант «Список сотрудников организации»
В пользовательских полях редактируем поле «Возраст интервалами» согласно заданию.
Для вывода количества используем существующий показатель «Количество».
Дале делаем гистограмму, в Точках делаем группировку Должность.Категория статистического учета (из раздела «Данные о рабочем месте»). В строках - «Возраст интервалами». В выводимые поля добавляем Количество. 
Задача2.
Создаем предопределенный вид расчета (ДОПОЛНИТЕЛЬНЫЙ): МатериальнаяВыгодаБассейн
Предопределенные показатели: СтажТочный, РыночнаяСтоимость (числовой, для сотрудника, ежемесячно перед расчетом зарплаты).
Убираем в формах запреты на работу с предопределенными элементами (быстрее, чем писать обработку по заполеннию).
Получение точного стажа: В модуле ПроведениеРасчетовПереопределяемый вносим правки:
· Функция ПолучитьСтруктуруНеобходимыхДанных:
в начале:
НеобходимыеДанные.Вставить("СтажТочный", Ложь);
в конце в цикле по показателям:
НеобходимыеДанные.СтажТочный = НеобходимыеДанные.СтажТочный Или Показатель = Справочники.ПоказателиСхемМотивации.СтажТочный;
меняем строку НеобходимыеДанные.Стаж.... на:
НеобходимыеДанные.Стаж = НеобходимыеДанные.Стаж Или Показатель = Справочники.ПоказателиСхемМотивации.Стаж Или Показатель = Справочники.ПоказателиСхемМотивации.СтажТочный;
· ПолучитьТекстПоказателя добавляем (для доп. начислений):
			?(НеобходимыеДанные.СтажТочный, "                 
			|	КОГДА Показатели.Показатель1 = ЗНАЧЕНИЕ(Справочник.ПоказателиСхемМотивации.СтажТочный)
			|		ТОГДА РАЗНОСТЬДАТ(НачалоПериода(ДанныеОСтаже.Период,Месяц), НачалоПериода(Основной.БазовыйПериодКонец,Месяц), МЕСЯЦ)+Выбор когда День(Основной.БазовыйПериодКонец)<День(ДанныеОСтаже.Период) Тогда -1 Иначе 0 конец", "") + 
В регистр расчета доп. нач. добавляем реквизит Номер (понадобится для свертки; хотя можно и без него).
Создаем наш документ:
Шапка: Организация, ПериодРегистрации, РыночнаяСтоимость.
ТЧ ГрафикПосещений: Сотрудник, День1..День10, Результат, ОплаченоДнейЧасов
На форме делаем кнопку «рассчитать», по которой:
	ЗаполнитьРассчитатьДвижения(Истина); 
	ЗаполнитьСумму(Движения.ДополнительныеНачисленияРаботниковОрганизаций);
В модуле документа:
Функция СформироватьЗапросПоТЧ()
	Запрос = Новый Запрос;
	Запрос.Текст = "ВЫБРАТЬ
	|	ПосещениеБассейнаСотрудники.Сотрудник КАК Сотрудник,
	|	ПосещениеБассейнаСотрудники.День1,
	|	ПосещениеБассейнаСотрудники.День2,
	|	ПосещениеБассейнаСотрудники.День3,
	|	ПосещениеБассейнаСотрудники.День4,
	|	ПосещениеБассейнаСотрудники.День5,
	|	ПосещениеБассейнаСотрудники.День6,
	|	ПосещениеБассейнаСотрудники.День7,
	|	ПосещениеБассейнаСотрудники.День8,
	|	ПосещениеБассейнаСотрудники.День9,
	|	ПосещениеБассейнаСотрудники.День10,
	|	ПосещениеБассейнаСотрудники.НомерСтроки,
	|	ПосещениеБассейнаСотрудники.Результат,
	|	ПосещениеБассейнаСотрудники.ОплаченоДнейЧасов,
	|	ВЫБОР
	|		КОГДА РаботникиОрганизации.ПериодЗавершения <= &ДатаСреза
	|				И РаботникиОрганизации.ПериодЗавершения <> ДАТАВРЕМЯ(1, 1, 1, 0, 0, 0)
	|			ТОГДА РаботникиОрганизации.ПодразделениеОрганизацииЗавершения
	|		ИНАЧЕ РаботникиОрганизации.ПодразделениеОрганизации
	|	КОНЕЦ КАК ПодразделениеОрганизации
	|ИЗ
	|	Документ.ПосещениеБассейна.ГрафикПосещений КАК ПосещениеБассейнаСотрудники
	|		ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.РаботникиОрганизаций.СрезПоследних(
	|				&ДатаСреза,
	|				Сотрудник В
	|					(ВЫБРАТЬ РАЗЛИЧНЫЕ
	|						Сотрудники.Сотрудник
	|					ИЗ
	|						Документ.ПосещениеБассейна.ГрафикПосещений КАК Сотрудники
	|					ГДЕ
	|						Сотрудники.Ссылка = &Ссылка)) КАК РаботникиОрганизации
	|		ПО (РаботникиОрганизации.Сотрудник = ПосещениеБассейнаСотрудники.Сотрудник)
	|ГДЕ
	|	ПосещениеБассейнаСотрудники.Ссылка = &Ссылка";
	Запрос.УстановитьПараметр("ссылка",ссылка);
	Запрос.УстановитьПараметр("ДатаСреза",КонецМесяца(ПериодРегистрации));
	возврат Запрос.Выполнить().Выбрать();
КонецФункции


Процедура ДобавитьСтрокуДвижения(НаДату, Сотрудник, Номер,Авторасчет,НомерРыночнаяСтоимость,Результат,ОплаченоДнейЧасов,ПодразделениеОрганизации) 
	Движение = Движения.ДополнительныеНачисленияРаботниковОрганизаций.Добавить(); 
	Движение.ВидРасчета = ПланыВидовРасчета.ДополнительныеНачисленияОрганизаций.МатериальнаяВыгодаБассейн; 
	Движение.ПериодРегистрации = ПериодРегистрации; 
	Движение.Сотрудник = Сотрудник; 
	Движение.ФизЛицо = Сотрудник.ФизЛицо; 
	Движение.Организация = Организация; 
	Движение.ОбособленноеПодразделение = Организация; 
	Движение.БазовыйПериодКонец = НаДату; // используется стажем 
	Движение.Номер = Номер; 
	Движение["Показатель"+НомерРыночнаяСтоимость] = РыночнаяСтоимость; 
	Движение.ПодразделениеОрганизации = ПодразделениеОрганизации; 
	Движение.Авторасчет = Авторасчет; 
	Движение.Результат = Результат; 	
	Движение.ОплаченоДнейЧасов = ОплаченоДнейЧасов;
КонецПроцедуры 

Процедура ЗаполнитьРассчитатьДвижения(Авторасчет) экспорт 
	Записать();
	ВидРасчета = ПланыВидовРасчета.ДополнительныеНачисленияОрганизаций.МатериальнаяВыгодаБассейн;
	НомерРыночнаяСтоимость = ВидРасчета.Показатели.Найти(Справочники.ПоказателиСхемМотивации.РыночнаяСтоимость, "Показатель").НомерСтроки;
	
	Движения.ДополнительныеНачисленияРаботниковОрганизаций.Очистить(); 
	Движения.ДополнительныеНачисленияРаботниковОрганизаций.Записывать = Истина; 
	Движения.НДФЛСведенияОДоходах.Очистить(); 
	Движения.НДФЛСведенияОДоходах.Записывать = Истина; 
	ПервыйДень = НачалоМесяца(ПериодРегистрации); 
	ТекСотрудник = СформироватьЗапросПоТЧ();
	Если Авторасчет Тогда
		// вносим отдельные записи для каждого дня (чтобы стаж считался на день и точно вычислить мат. выгоду)
		Пока ТекСотрудник.Следующий() Цикл //Для каждого ТекСотрудник из ГрафикПосещений Цикл
			Для Колонка = 1 по 10 Цикл 
				Если ТекСотрудник["День"+Колонка] Тогда 
					ДобавитьСтрокуДвижения((ПервыйДень+86400*(Колонка-1)), ТекСотрудник.Сотрудник, ТекСотрудник.НомерСтроки,Авторасчет,НомерРыночнаяСтоимость,0,1,ТекСотрудник.ПодразделениеОрганизации); 
				КонецЕсли; 
			КонецЦикла; 
		КонецЦикла; 
	Иначе
		// суммарную запись берем из документа (одна на сотрудника):
		Пока ТекСотрудник.Следующий() Цикл//Для каждого ТекСотрудник из ГрафикПосещений Цикл 
			ДобавитьСтрокуДвижения(ПервыйДень, ТекСотрудник.Сотрудник, ТекСотрудник.НомерСтроки,Авторасчет,НомерРыночнаяСтоимость,ТекСотрудник.Результат,ТекСотрудник.ОплаченоДнейЧасов,ТекСотрудник.ПодразделениеОрганизации); 
		КонецЦикла; 
	КонецЕсли;
	ПроведениеРасчетовПереопределяемый.РассчитатьЗаписиРегистраРасчета("ДополнительныеНачисленияРаботниковОрганизаций", Движения.ДополнительныеНачисленияРаботниковОрганизаций, Движения.НДФЛСведенияОДоходах, Организация, Организация); 
КонецПроцедуры 

Процедура ОбработкаПроведения(Отказ, Режим)
	ЗаполнитьРассчитатьДвижения(Ложь); 
КонецПроцедуры 

Процедура ЗаполнитьСумму(ДвиженияНач)Экспорт 
	ТаблицаДвижений = ДвиженияНач.Выгрузить(, "Сотрудник, Номер, Результат, ОплаченоДнейЧасов"); 
	ТаблицаДвижений.Свернуть("Сотрудник, Номер", "Результат, ОплаченоДнейЧасов"); 
	Для Каждого ТекДвиж из ТаблицаДвижений Цикл 
		ГрафикПосещений[ТекДвиж.Номер-1].Результат = ТекДвиж.Результат; 
		ГрафикПосещений[ТекДвиж.Номер-1].ОплаченоДнейЧасов = ТекДвиж.ОплаченоДнейЧасов; 
	КонецЦикла; 
	Движения.ДополнительныеНачисленияРаботниковОрганизаций.Очистить(); 
	Движения.НДФЛСведенияОДоходах.Очистить(); 
	Движения.ДополнительныеНачисленияРаботниковОрганизаций.Записать(Ложь);
	Движения.НДФЛСведенияОДоходах.Записать();
КонецПроцедуры 
Заполняем вид расчета:
Способ расчета – произвольный. Формула расчета:
(1-ОценитьПо(СтажТочный,ШкалаЛьготы))*РыночнаяСтоимость 
Является доходом в натуральной форме, НДФЛ – наверно код 2530 (или 2630?), Взносы и ЕСН –облагается вроде бы.
В шкале обязательно учесть, что значение «с» учитывает по «>», а «по» - «<=».
Создаем пару сотрудников, заполняем пару документов по бассейну, создаем за те же месяцы документы НачисленияЗП (чтобы НДФЛ посчитался). Все.
Отчет: 
ВЫБРАТЬ
	ДополнительныеНачисленияРаботниковОрганизаций.Сотрудник КАК Сотрудник,
	ДополнительныеНачисленияРаботниковОрганизаций.ПодразделениеОрганизации КАК ПодразделениеОрганизации,
	ДополнительныеНачисленияРаботниковОрганизаций.Результат КАК Результат,
	ДополнительныеНачисленияРаботниковОрганизаций.ПериодРегистрации КАК ПериодРегистрации,
	ДополнительныеНачисленияРаботниковОрганизаций.ОплаченоДнейЧасов КАК ПосещенийБассейна
ИЗ
	РегистрРасчета.ДополнительныеНачисленияРаботниковОрганизаций КАК ДополнительныеНачисленияРаботниковОрганизаций
ГДЕ
	ДополнительныеНачисленияРаботниковОрганизаций.Организация = &Организация
	И ДополнительныеНачисленияРаботниковОрганизаций.ВидРасчета = Значение(ПланВидовРасчета.ДополнительныеНачисленияОрганизаций.МатериальнаяВыгодаБассейн)
	И ДополнительныеНачисленияРаботниковОрганизаций.ПериодРегистрации МЕЖДУ &ДатаНач И &ДатаКон

Задача1: График роста ЗП работников за первый квартал
Принимаем сотрудников с 1.01.2012, с 1.03 вводим «Ввод постоянного начисления или удержания организации». Начисляем ЗП за три месяца.
Берем отчет «Анализ начислений сотрудникам организаций».
В расширенных настройках вызываем конструктор,  в поля добавляем Начислено, в Точки Период регистрации (в «условном оформлении» для отчета в целом не забыть поменять формат оформления поля), в Серии – Организация. Тип диаграммы – График. Радуемся жизни.
Задача2. 
ПлановыеНачисления или ГрафикиРаботПовидамВремени – ПРАВИТЬ НЕЛЬЗЯ и не нужно. Снизят минимум на 2 балла. Причина:  данные по тому, что охраняли, согласно условиям задачи не влияют на начисления. Работник получает по тарифной ставке за отработанное время – ВСЕ. Потому все что нужно – это получить в итоге красивый отчет.
Создаем спр. ОбъектыОхраны
Регистр сведений: ОхранаОбъектов (запись регистратором); Измерения: ДатаОхраны, Организация, Сотрудник; Ресурсы: ОбъектОхраны.
Документ: «ГрафикОхраныОбъектов» и отчет. Выносим их в интерфейс.
Реквизиты документа: Организация, ПериодРегистрации.
ТЧ ГрафикОхраны: ОбъектОхраны, День1-День31 (спр. СотрудникиОрганизаций).
Обработка проведения (написана исходя из того, что в днях были при решении объекты охраны, а не сотрудники – косяк был при решении. Потому нужно поправить, что несложно):
	Движения.ОхранаОбъектов.Записывать = Истина;
	Движения.ОхранаОбъектов.Очистить();
	НачалоМесяцаРегистрации = НачалоМесяца(ПериодРегистрации);
	СекундВДне = 24*60*60;
	Для Каждого ТекСтрокаГрафикОхраны Из ГрафикОхраны Цикл
		Для сч=1 по 31 Цикл
			ОбъектОхраны = ТекСтрокаГрафикОхраны["День"+сч];
			Если ЗначениеЗаполнено(ОбъектОхраны) Тогда
				Движение = Движения.ОхранаОбъектов.Добавить();
				Движение.ДатаОхраны = НачалоМесяцаРегистрации+(сч-1)*СекундВДне;
				Движение.Организация = Организация;
				Движение.Сотрудник = ТекСтрокаГрафикОхраны.Сотрудник;
				Движение.ОбъектОхраны = ОбъектОхраны;
			КонецЕсли;
		КонецЦикла;
	КонецЦикла;
Задание простое, потому усложним отчет, чтобы не облажаться. Учтем, что график может быть заполнен не полностью, и для таких дней сумму начислений отнесем на неизвестный объект:
ВЫБРАТЬ
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.ПериодРегистрации,
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.ОплаченоДнейЧасов,
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.Результат,
	ОхранаОбъектов.ОбъектОхраны,
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.Показатель1 * КОЛИЧЕСТВО(ОхранаОбъектов.ДатаОхраны) КАК НачисленоЗаОхрануОбъектов,
	КОЛИЧЕСТВО(ОхранаОбъектов.ДатаОхраны) КАК ОтработаноДней,
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.Сотрудник,
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.Организация,
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.ПодразделениеОрганизации
ПОМЕСТИТЬ ВТНачисленияОбщие
ИЗ
	РегистрРасчета.ОсновныеНачисленияРаботниковОрганизаций.ФактическийПериодДействия КАК ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия
		ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ОхранаОбъектов КАК ОхранаОбъектов
		ПО (ОхранаОбъектов.ДатаОхраны МЕЖДУ ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.ПериодДействияНачало И ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.ПериодДействияКонец)
			И (ОхранаОбъектов.Организация = ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.Организация)
			И (ОхранаОбъектов.Сотрудник = ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.Сотрудник)
ГДЕ
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.ВидРасчета = ЗНАЧЕНИЕ(ПланВидовРасчета.ОсновныеНачисленияОрганизаций.ТарифДневной)

СГРУППИРОВАТЬ ПО
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.ПериодРегистрации,
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.ОплаченоДнейЧасов,
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.Результат,
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.Показатель1,
	ОхранаОбъектов.ОбъектОхраны,
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.Сотрудник,
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.Организация,
	ОсновныеНачисленияРаботниковОрганизацийФактическийПериодДействия.ПодразделениеОрганизации
;
////////////////////////////////////////////////////////////////////////////////
ВЫБРАТЬ
	ВТНачисленияОбщие.ПериодРегистрации,
	ВТНачисленияОбщие.ОплаченоДнейЧасов,
	ВТНачисленияОбщие.Результат,
	СУММА(ВТНачисленияОбщие.НачисленоЗаОхрануОбъектов) КАК НачисленоЗаОхрануОбъектов,
	СУММА(ВТНачисленияОбщие.ОтработаноДней) КАК ОтработаноДней,
	ВТНачисленияОбщие.Сотрудник,
	ВТНачисленияОбщие.Организация,
	ВТНачисленияОбщие.ПодразделениеОрганизации
ПОМЕСТИТЬ ВТИтогПоСотрудникам
ИЗ
	ВТНачисленияОбщие КАК ВТНачисленияОбщие

СГРУППИРОВАТЬ ПО
	ВТНачисленияОбщие.ПериодРегистрации,
	ВТНачисленияОбщие.Сотрудник,
	ВТНачисленияОбщие.Организация,
	ВТНачисленияОбщие.ПодразделениеОрганизации,
	ВТНачисленияОбщие.ОплаченоДнейЧасов,
	ВТНачисленияОбщие.Результат
;
////////////////////////////////////////////////////////////////////////////////
ВЫБРАТЬ
	ВТНачисленияОбщие.ПериодРегистрации,
	ВТНачисленияОбщие.ОбъектОхраны,
	ВТНачисленияОбщие.Сотрудник,
	ВТНачисленияОбщие.Организация,
	ВТНачисленияОбщие.ПодразделениеОрганизации,
	ВТНачисленияОбщие.НачисленоЗаОхрануОбъектов,
	ВТНачисленияОбщие.ОтработаноДней
ИЗ
	ВТНачисленияОбщие КАК ВТНачисленияОбщие
ОБЪЕДИНИТЬ ВСЕ
ВЫБРАТЬ
	ВТИтогПоСотрудникам.ПериодРегистрации,
	null,
	ВТИтогПоСотрудникам.Сотрудник,
	ВТИтогПоСотрудникам.Организация,
	ВТИтогПоСотрудникам.ПодразделениеОрганизации,
	ВТИтогПоСотрудникам.Результат-ВТИтогПоСотрудникам.НачисленоЗаОхрануОбъектов,
	ВТИтогПоСотрудникам.ОплаченоДнейЧасов-ВТИтогПоСотрудникам.ОтработаноДней
ИЗ
	ВТИтогПоСотрудникам КАК ВТИтогПоСотрудникам 
В форме можно навести украшательство, чтобы компенсировать что задача простая - можно взять из ввода индивидуальных графиков к примеру.


[bookmark: _Toc349662376]Билет 8. График охраны физ. лиц (ставка зависит от того, кого охраняем).
Задача1: Отчет о суммах облагаемого дохода физ. лиц.
При принятии сотрудника с авторским доходом: в доп. начислении, которое указывается в договоре, нужно выбрать правильный код дохода, как в задании.
Суммы Вычетов с кодом 403 сотрудников-договорняков вводим в документе НЗРО на закладке договоров – считает дальше верно.
Универсальный отчет по «НДФЛ сведения о доходах». Создаем пользовательское поле «Выражение»: [Сумма дохода] - [Сумма вычета]; Итоги: Сумма([Сумма дохода] - [Сумма вычета]). Ну и выводим как на рисунке.
Задача2.  ПРИМЕЧАНИЕ ПО РЕШЕНИЮ: лучше ставку тарифа вычислять в момент расчета по объекту охраны с помощью нового предопределенного показателя. Это если время останется.
По сути есть несколько путей решения, и все кривые. Пойдем по тому, по которому есть отзывы, что прокатил на экзамене. Обосновать при необходимости можно.
Лучше сразу создать свой предопред. вид расчета, по образце дневного тарифа. Цель: чтобы не придрались, что отчет отделяем криво по объекту.
Создаем спр. ОбъектыОхраны
Регистр сведений: ОхранаОбъектов (запись регистратором); Измерения: ДатаОхраны, Организация, Сотрудник; Ресурсы: ОбъектОхраны.
Регистр сведений: СтавкиОхраныОбъектов (независимый; период месяц); Измерения: ОбъектОхраны; Ресурсы: ДневнаяСтавка.
Документ: «ГрафикОхраныОбъектов» и отчет. Выносим их в интерфейс.
Реквизиты документа: Организация, ПериодРегистрации.
ТЧ ГрафикОхраны: ОбъектОхраны, День1-День31 (спр. СотрудникиОрганизаций).
В движениях добавим регистр сведений плановых начислений работникам организаций и составим модуль проведения:
	// для оптимизации сгруппируем идущие вподряд дни по одному объекту охраны с одним и тем же работником
	// это допустимо с учетом того, что ставки согласно тек. настройкам задаются раз в месяц. 
	// Если чаще - то просто не нужно групировать.
	ОхранаСотрудникамиПоДням = новый ТаблицаЗначений;
	ОхранаСотрудникамиПоДням.Колонки.Добавить("ОбъектОхраны");
	ОхранаСотрудникамиПоДням.Колонки.Добавить("Сотрудник");
	ОхранаСотрудникамиПоДням.Колонки.Добавить("ДатаОхраны");
	НачалоМесяцаРегистрации = НачалоМесяца(ПериодРегистрации);
	СекундВДне = 24*60*60;
	Для Каждого ТекСтрокаГрафикОхраны Из ГрафикОхраны Цикл
		Для сч=1 по 31 Цикл
			Сотрудник = ТекСтрокаГрафикОхраны["День"+сч];
			Если ЗначениеЗаполнено(Сотрудник) Тогда
				СтрокаТЗ = ОхранаСотрудникамиПоДням.Добавить();
				СтрокаТЗ.ДатаОхраны = НачалоМесяцаРегистрации+(сч-1)*СекундВДне;
				СтрокаТЗ.Сотрудник = Сотрудник;
				СтрокаТЗ.ОбъектОхраны = ТекСтрокаГрафикОхраны.ОбъектОхраны;
			КонецЕсли;
		КонецЦикла;
	КонецЦикла;
	МассивОбъектыОхраны = ОхранаСотрудникамиПоДням.ВыгрузитьКолонку("ОбъектОхраны");
	Запрос = новый Запрос("ВЫБРАТЬ
	                      |	СтавкиОхраныОбъектовСрезПоследних.ОбъектОхраны,
	                      |	СтавкиОхраныОбъектовСрезПоследних.ДневнаяСтавка
	                      |ИЗ
	                      |	РегистрСведений.СтавкиОхраныОбъектов.СрезПоследних(&НаДату, ОбъектОхраны В (&ОбъектОхраны)) КАК СтавкиОхраныОбъектовСрезПоследних");
	Запрос.УстановитьПараметр("ОбъектОхраны",МассивОбъектыОхраны);
	Запрос.УстановитьПараметр("НаДату",НачалоМесяцаРегистрации);
	Выборка = Запрос.Выполнить().Выбрать();
	ДневныеСтавкипоОбъектам = новый Соответствие;
	Пока Выборка.Следующий() Цикл
		ДневныеСтавкипоОбъектам.Вставить(Выборка.ОбъектОхраны,Выборка.ДневнаяСтавка);
	Конеццикла;
	
	ОхранаСотрудникамиПоДням.Сортировать("Сотрудник Возр, ДатаОхраны Возр");
	ВидРасчета = ПланыВидовРасчета.ОсновныеНачисленияОрганизаций.ТарифДневнойПоОбъектамОхраны;
	// заполняем движения              
	// заодно проверим случаи, когда одни сотрудник прописан на одну дату на разные объекты
	ПредыдущийСотрудник = неопределено;
	ПредыдущийОбъектОхраны = неопределено;
	ПредыдущаяДатаОхраны = '00010101';
	
	Движения.ПлановыеНачисленияРаботниковОрганизаций.Записывать = Истина;
	Движения.ПлановыеНачисленияРаботниковОрганизаций.Очистить();
	Для Каждого ТекСтрокаГрафик Из ОхранаСотрудникамиПоДням Цикл
		Сотрудник = ТекСтрокаГрафик.Сотрудник;
		ДатаОхраны = ТекСтрокаГрафик.ДатаОхраны;
		ОбъектОхраны = ТекСтрокаГрафик.ОбъектОхраны;
		Если ПредыдущийСотрудник = Сотрудник 
			и ПредыдущаяДатаОхраны = ДатаОхраны Тогда
			ОбщегоНазначения.СообщитьОбОшибке("Сотрудник "+Сотрудник+" указан более чем на одном объекте охраны на дату "+ДатаОхраны,Отказ);
		КонецЕсли;
		Если Сотрудник<>ПредыдущийСотрудник Тогда
			ПредыдущийСотрудник = Сотрудник;
			ПредыдущийОбъектОхраны = неопределено;
		КонецЕсли;
		Если ПредыдущаяДатаОхраны+СекундВДне<>ДатаОхраны Тогда
			ПредыдущийОбъектОхраны = неопределено;
		КонецЕсли;
		ПредыдущаяДатаОхраны = ДатаОхраны;
		Если ОбъектОхраны <> ПредыдущийОбъектОхраны Тогда
			ПредыдущийОбъектОхраны = ОбъектОхраны;
			// добавляем движение, сразу указывая ставку. 
			Движение = Движения.ПлановыеНачисленияРаботниковОрганизаций.Добавить();
			Движение.Сотрудник = Сотрудник;
			Движение.Организация = Организация;
			Движение.ВидРасчетаИзмерение = ВидРасчета;
			Движение.ВидРасчета = ВидРасчета;
			Движение.ОбъектОхраны = ОбъектОхраны;
			Движение.Показатель1 = ДневныеСтавкиПоОбъектам[ОбъектОхраны];
			
			Движение.Действие = Перечисления.ВидыДействияСНачислением.Начать;
			Движение.ДействиеЗавершения = Перечисления.ВидыДействияСНачислением.Прекратить;
			Движение.Период = ДатаОхраны;
			Движение.ПериодЗавершения = ДатаОхраны + СекундВДне;// по умолчанию считаем, что в этот день и закончится
		Иначе
			// продлим предыдущее, т.к. сотрудник и объект охраны не изменились.
			// даты, в которые
			Движение.ПериодЗавершения = ДатаОхраны + СекундВДне;
		КонецЕсли;
	КонецЦикла;
Открываем модуль документа «Начисление зарплаты работникам организаций». Ищем строки (в ВыполнитьАвтозаполнение):
Запрос.Текст = ОсновнойТекстЗапроса;
РезультатЗапроса = Запрос.Выполнить();
Нам нужны первые из всех найденных. Выше идёт интересующий нас текст запроса. Перед блоком «УПОРЯДОЧИТЬ ПО», где заканчиваются соединения, добавим наше соединение:
ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ПлановыеНачисленияРаботниковОрганизаций КАК ОбъектыОхраны
	ПО СписокНачислений.Сотрудник = ОбъектыОхраны.Сотрудник
		И ОбъектыОхраны.Организация = &парамОрганизация
		И СписокНачислений.Период = ОбъектыОхраны.Период
		И СписокНачислений.ВидРасчета = ОбъектыОхраны.ВидРасчета
На практике, конечно, лучше внести это корректно по цепочке запросов выше. Если время будет.
В списке выборки добавим поле «ОбъектыОхраны.ОбъектОхраны». Теперь опустимся чуть ниже строк, что мы искали. Ищем код «СтрокаПолейОсновныхНачислений="…» и дописываем в строчку «, ОбъектОхраны».
Поднимаемся выше по тексту. Там видим код «ТЗНачисления = СформироватьТаблицуНачислений(…». Переходим  к фукнции «СформироватьТаблицуНачислений». Первым делом добавляем в таблицу начислений новую колонку (куда добавить, догадаетесь сами):
ТЗНачисления.Колонки.Добавить("ОбъектОхраны ");

Ищем строку «// Добавим новую строку начислений» и после добавления строки вставляем код:
НоваяСтрока.ОбъектОхраны = Начисления.ОбъектОхраны;
Заполнение документа готово. Приступим к расчёту. Ищем функцию СформироватьЗапросПоНачисления. В ней есть куча запросов, нас интересуют только 2 – первый и последний. В первом в поле выборки добавляем реквизит ОбъектОхраны, во втором – его же в выборку из временной таблицы, полученной в первом запросе. Потом идёи в процедуру ДобавитьСтрокуОсновныхНачислений и туда дописываем текст:
Движение.ОбъектОхраны = ВыборкаПоСтрокамДокумента.ОбъектОхраны;
Расчёт готов!
Теперь добавим пустой отчёт «Вознаграждение работников» и вынесем объекты в интерфейс.
Запускаем предприятие, включаем управленческий учёт, вводим данные об организации, производственный календарь, графики работы, принимаем на работу сотрудников. Потом заводим объекты охраны и устанавливаем сумму оплаты за объект. После этого вводим график охраны и производим расчёт.
Теперь по готовым данным отчёт составить проще простого. Вот мой текст запроса:
ВЫБРАТЬ
	ОсновныеНачисленияРаботниковОрганизаций.Сотрудник,
	ОсновныеНачисленияРаботниковОрганизаций.Результат,
	ОсновныеНачисленияРаботниковОрганизаций.ОплаченоДнейЧасов,
	ОсновныеНачисленияРаботниковОрганизаций.ОбъектОхраны,
	ОсновныеНачисленияРаботниковОрганизаций.ПериодРегистрации
ИЗ
	РегистрРасчета.ОсновныеНачисленияРаботниковОрганизаций КАК ОсновныеНачисленияРаботниковОрганизаций
ГДЕ
	ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета = &ВидРасчета
	И ОсновныеНачисленияРаботниковОрганизаций.ОбъектОхраны <> &ОбъектОхраны
ВидРасчета – наш тариф по дням, ОбъектОхраны – пустая ссылка справочника объектов охраны


[bookmark: _Toc349662377]Билет 9. Надбавка за квалификацию (сертификат).
Задача1: Средняя численность сотрудников
Берем одноименный отчет, серии ФизЛицо.Пол, Точки – Месяц, выводим общее кол-во. Строится криво, но работает.
Задача1: Добавляем справочники ВидыАттестатов, ПрограммныеПродукты.
РС НадбавкаЗаКвалификацию (периодичность месяц): измерение ВидАттестата; ресурс Процент
В РС Плановые насчиления добавляем измерения ВидАттестата, ПрограммныйПродукт. В РР ОсновныеНачисленияОрганизации добавляем реквизиты ВидАттестата, ПрограммныйПродукт. Почему в Основные: хоть дополнительные начисления и можно писать в плановые, но похоже они скоро умрут. Да и не используются они как плановые стандартными документами, а потому и нам использовать не есть гуд.
Добавляем предопределенное основноеНачислениеОрганизации: ДоплатаЗаКвалификацию (в форме снимаем блокировку редактирования предопределенных – для быстроты, чем писать заполнение реквизитов самому).
Документ АттестацияРаботников. Реквизиты: Организация; ТЧ Сотрудники: Сотрудник, ВидАттестата, ПрограммныйПродукт. Обработка проведения:
	Движения.ПлановыеНачисленияРаботниковОрганизаций.Записывать = Истина;
	Движения.ПлановыеНачисленияРаботниковОрганизаций.Очистить();
	ВидРасчета = ПланыВидовРасчета.ОсновныеНачисленияОрганизаций.НайтиПоКоду("DOP");
	//ВидРасчета = ПланыВидовРасчета.ОсновныеНачисленияОрганизаций.ДоплатаЗаКвалификацию;
	Для Каждого ТекСтрокаСотрудники Из Сотрудники Цикл
		Движение = Движения.ПлановыеНачисленияРаботниковОрганизаций.Добавить();
		Движение.Сотрудник = ТекСтрокаСотрудники.Сотрудник;
		Движение.Организация = Организация;
		Движение.ВидРасчетаИзмерение = ВидРасчета;
		Движение.ВидРасчета = ВидРасчета;
		Движение.Действие = Перечисления.ВидыДействияСНачислением.Начать;
		Движение.Период = Дата;
		Движение.ВидАттестата = ТекСтрокаСотрудники.ВидАттестата;
		Движение.ПрограммныйПродукт = ТекСтрокаСотрудники.ПрограммныйПродукт;
	КонецЦикла; 
Изменения документа НачислениеЗарплатыРаботникамОрганизаций. Добавляем реквизиты: в ТЧ начисления: ВидАттестата, ПрограммныйПродукт. 
Ищем строки в модуле объекта (в ВыполнитьАвтозаполнение):
Запрос.Текст = ОсновнойТекстЗапроса;
РезультатЗапроса = Запрос.Выполнить();
Нам нужны первые из всех найденных. Выше идёт интересующий нас текст запроса. Правим его, добавляя получение полей ВидАттестата, ПрограммныйПродукт. Дело муторное, на всякий случай ключевые места (вторая строка – где вставляем):
1) 		|,НачисленияРаботниковОрганизации.ВидАттестата,НачисленияРаботниковОрганизации.ПрограммныйПродукт//АХТУНГ
		|ПОМЕСТИТЬ ВТРаботникиИНачисленияПлановыеНачисления
2) 3 штуки:		|,НачисленияРаботниковОрганизации.ВидАттестата,НачисленияРаботниковОрганизации.ПрограммныйПродукт//АХТУНГ
		|		ИЗ
		|			РегистрСведений.ПлановыеНачисленияРаботниковОрганизаций……
3) 		|,ПлановыеНачисления.ВидАттестата,ПлановыеНачисления.ПрограммныйПродукт//АХТУНГ
		|ПОМЕСТИТЬ ВТПлановыеНачисленияНадбавкиПоШР
4) 		|,null,null//АХТУНГ
		|ИЗ
		|	ВТНачисленияПоШР КАК НачисленияПоШР
5) 		|,НачисленияНадбавки.ВидАттестата,НачисленияНадбавки.ПрограммныйПродукт//АХТУНГ
		|ПОМЕСТИТЬ ВТРаботникиИНачисления
6) 		|,НачисленияНадбавки.ВидАттестата,НачисленияНадбавки.ПрограммныйПродукт//АХТУНГ
		|ИЗ
		|	ВТПлановыеНачисленияНадбавкиПоШР КАК НачисленияНадбавки
7) 		|,СписокНачислений.ВидАттестата,СписокНачислений.ПрограммныйПродукт//АХТУНГ
		|ПОМЕСТИТЬ ВТРаботникиИНачисленияПоСхемамМотивации
8)Финальный: 	|,СписокНачислений.ВидАттестата,СписокНачислений.ПрограммныйПродукт//АХТУНГ
		|ИЗ
		|	ВТРаботникиИНачисленияПоСхемамМотивации КАК СписокНачислений
Тут же поиском находим команду «СтрокаПолейОсновныхНачислений= “…..» и дописываем в конец «, ВидАттестата, ПрограммныйПродукт».
Теперь нам необходимо поправить функцию СформироватьТаблицуНачислений:
ТЗНачисления.Колонки.Добавить("ВидАттестата");//АХТУНГ
ТЗНачисления.Колонки.Добавить("ПрограммныйПродукт");//АХТУНГ 
Затем ищем строчку «НоваяСтрока = ТЗНачисления.Добавить();» и ниже дописываем:
НоваяСтрока.ВидАттестата = Начисления.ВидАттестата; //АХТУНГ
НоваяСтрока.ПрограммныйПродукт = Начисления.ПрограммныйПродукт; //АХТУНГ 
В этой же функции выше ищем «НайтиСредиНезакрытых(…..». В вызове функции дописываем в конец параметр «, Начисления.ВидАттестата» и находим эту функцию. Правим в ней объявление – добавляем параметры «Начисления.ВидАттестата,Начисления.ПрограммныйПродукт» и условие в цикле:
Если (ПоискОсновного И НезакрытыеСтроки[Индекс].Основное) 
	ИЛИ (НЕ ПоискОсновного И НезакрытыеСтроки[Индекс].ВидРасчета = ВидРасчета
//!!!!!
И НезакрытыеСтроки[Индекс].ВидАттестата = ВидАттестата
	И НезакрытыеСтроки[Индекс].ПрограммныйПродукт = ПрограммныйПродукт
			) Тогда 
Мы поправили заполнение табличной части. Теперь нам надо поправить заполнение набора записей регистра расчёта. Для этого нам нужно подкорректировать функцию СформироватьЗапросПоНачисления: в текст запроса просто-напросто добавим поля ВидАттестата и ПрограммныйПродукт (первый и последний запрос). И добавим в ДобавитьСтрокуОсновныхНачислений эти же поля.
Теперь можно переходить непосредственно к процедурам расчёта.
Добавляем предопределенный Показатель Схем Мотивации:ПроцентДоплатыЗаКвалификацию
В общем модуле ПроведениеРасчетовПереопределяемый в функции ПолучитьСтруктуруНеобходимыхДанных добавляем показатель ПроцентДоплатыЗаКвалификацию.
В ПолучитьДанныеДляРасчета добавляем получение данных для показателя:
Если НеобходимыеДанные.ПроцентДоплатыЗаКвалификацию Тогда//АХТУНГ
	ТекстЗапроса = ТекстЗапроса + "
	|ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.НадбавкаЗаКвалификацию.СрезПоследних(&ПериодРегистрации) КАК НадбавкаЗаКвалификацию			
	|ПО НадбавкаЗаКвалификацию.ВидАттестата = Основной.ВидАттестата";
	Запрос.УстановитьПараметр("ПериодРегистрации",	ПериодРегистрации);
КонецЕсли;
В ПолучитьТекстПоказателя. В условие:
НеобходимыеДанные. ПроцентДоплатыЗаКвалификацию или
И по тексту:
?(НеобходимыеДанные.ПроцентДоплатыЗаКвалификацию, "                 
|	КОГДА Показатели.Показатель1 = ЗНАЧЕНИЕ(Справочник.ПоказателиСхемМотивации.ПроцентДоплатыЗаКвалификацию)
|		ТОГДА ЕстьNull(НадбавкаЗаКвалификацию.Процент,0)", "") +
В режиме 1С:Предприятие создаем вид расчета с кодом DOP и формулой:
РасчетнаяБаза*ПроцентДоплатыЗаКвалификацию/100
Заполняем расчетную базу и остальные поля.
Отчет (число аттестатов вычисляем по разным прог. продуктам.):
ВЫБРАТЬ
	ОсновныеНачисленияРаботниковОрганизаций.ПериодРегистрации КАК ПериодРегистрации,
	СУММА(ОсновныеНачисленияРаботниковОрганизаций.Результат) КАК НадбавкаЗаКвалификацию,
	ОсновныеНачисленияРаботниковОрганизаций.Сотрудник КАК Сотрудник,
	ОсновныеНачисленияРаботниковОрганизаций.ВидАттестата КАК ВидАттестата,
	ОсновныеНачисленияРаботниковОрганизаций.Организация КАК Организация,
	КОЛИЧЕСТВО(РАЗЛИЧНЫЕ ОсновныеНачисленияРаботниковОрганизаций.ПрограммныйПродукт) КАК КоличествоАттестатов
ИЗ
	РегистрРасчета.ОсновныеНачисленияРаботниковОрганизаций КАК ОсновныеНачисленияРаботниковОрганизаций
ГДЕ
	ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета = &ВидРасчета
	И ОсновныеНачисленияРаботниковОрганизаций.ПериодРегистрации МЕЖДУ &ДатаНач И &ДатаКон
СГРУППИРОВАТЬ ПО
	ОсновныеНачисленияРаботниковОрганизаций.Сотрудник,
	ОсновныеНачисленияРаботниковОрганизаций.ВидАттестата,
	ОсновныеНачисленияРаботниковОрганизаций.ПериодРегистрации,
	ОсновныеНачисленияРаботниковОрганизаций.Организация


[bookmark: _Toc349662378]Билет 10. Вознаграждение процентом от объема продаж своих работников.
Задача1:Отчет по количеству работников по значению стажа работы в отрасли.
Заводим работников и принимаем на работу. В справочнике ФизЛиц в Трудовой деятельности вводим значения стажа (и сам вид стажа создаем).
Назначаем постоянное начисление по надбавке (от стажа процентом, указываем наш стаж и заполняем шкалу с процентами; стаж в шкале должен быть указан в месяцах, значения - проценты).
Отчет «Списки сотрудников организации»,  за основу вариант «Стаж работы». В расширенной настройке создаем пользовательское поле «Стаж интервалами» (по аналогии с возраст интервалами). Указываем интервалы с отбором по виду стажа, а также числу лет стажа. Ставим в целом для отчета отбор по виду стажа. Тип отчета – таблица. Поля: количество. Строки: должность.Категория статистического учета. Столбцы: Стаж интервалами.
Задача2: .
Создаем оборотный регистр ПродажиРаботников: Организация, сотрудник;  Сумма
Создаем предопределенный показатель схем мотивации: СуммаПродаж
Создаем документ ДанныеОбОбъёмахПродаж: Организация, ПериодРегистрации.  ТЧ Сотрудники: Сотрудник, День1-День11 (число(15,2)). При проведении реализуем перерасчет, если затронуты документы начисления:
// регистр ПродажиРаботников 
Движения.ПродажиРаботников.Записывать = Истина;
Движения.ПродажиРаботников.Очистить();
СекундВДне=24*60*60;
Для Каждого ТекСтрокаСотрудники Из Сотрудники Цикл
	Для сч=1 по 11 Цикл
		Сумма = ТекСтрокаСотрудники["День"+сч];
		Если Сумма>0 Тогда
			Движение = Движения.ПродажиРаботников.Добавить();
			Движение.Период = ПериодРегистрации + (сч-1)*СекундВДне;
			Движение.Организация = Организация;
			Движение.Сотрудник = ТекСтрокаСотрудники.Сотрудник;
			Движение.Сумма = Сумма;
		КонецЕсли;
	КонецЦикла;
КонецЦикла;

// перерасчет
Движения.Записать();
//Запрос по записям регистра расчета
Запрос = Новый Запрос("ВЫБРАТЬ РАЗЛИЧНЫЕ
|	ОсновныеНачисленияОрганизаций.Ссылка
|ПОМЕСТИТЬ ВТНачисленияУчитывающиеСуммуПродаж
|ИЗ
|	ПланВидовРасчета.ОсновныеНачисленияОрганизаций.Показатели КАК ОсновныеНачисленияОрганизаций
|ГДЕ
|	ОсновныеНачисленияОрганизаций.Показатель = ЗНАЧЕНИЕ(Справочник.ПоказателиСхемМотивации.СуммаПродаж)
|;
|////////////////////////////////////////////////////////////////////////////////
|ВЫБРАТЬ РАЗЛИЧНЫЕ
|	ПродажиРаботников.Сотрудник,
|	ПродажиРаботников.Организация,
|	ОсновныеНачисленияРаботниковОрганизаций.ФизЛицо КАК ФизЛицо,
|	ОсновныеНачисленияРаботниковОрганизаций.Регистратор КАК Регистратор
|ИЗ
|	РегистрНакопления.ПродажиРаботников КАК ПродажиРаботников
|		ЛЕВОЕ СОЕДИНЕНИЕ РегистрРасчета.ОсновныеНачисленияРаботниковОрганизаций КАК ОсновныеНачисленияРаботниковОрганизаций
|		ПО ПродажиРаботников.Сотрудник = ОсновныеНачисленияРаботниковОрганизаций.Сотрудник
|			И ПродажиРаботников.Организация = ОсновныеНачисленияРаботниковОрганизаций.Организация
|			И (ПродажиРаботников.Период МЕЖДУ ОсновныеНачисленияРаботниковОрганизаций.ПериодРегистрации И КОНЕЦПЕРИОДА(ОсновныеНачисленияРаботниковОрганизаций.ПериодРегистрации, МЕСЯЦ))
|ГДЕ
|	ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета В
|			(ВЫБРАТЬ
|				ВТНачисленияУчитывающиеСуммуПродаж.Ссылка
|			ИЗ
|				ВТНачисленияУчитывающиеСуммуПродаж КАК ВТНачисленияУчитывающиеСуммуПродаж)
|	И ПродажиРаботников.Регистратор = &Регистратор
|ИТОГИ ПО
|	Регистратор");
Запрос.УстановитьПараметр("Регистратор", Ссылка);
ВыборкаДокументов = Запрос.Выполнить().Выбрать(ОбходРезультатаЗапроса.ПоГруппировкам);
Перерасчет = РегистрыРасчета.ОсновныеНачисленияРаботниковОрганизаций.Перерасчеты.ПерерасчетОсновныхНачислений;
Пока ВыборкаДокументов.Следующий() Цикл
	//Создание набора записей
	НаборЗаписей = Перерасчет.СоздатьНаборЗаписей();
	НаборЗаписей.Отбор.ОбъектПерерасчета.Значение = ВыборкаДокументов.Регистратор;
	
	Выборка = ВыборкаДокументов.Выбрать();
	//Создание записей перерасчета
	Пока Выборка.Следующий() Цикл
		Запись = НаборЗаписей.Добавить();
		Запись.Сотрудник = Выборка.Сотрудник;
		Запись.Организация = Выборка.Организация;
		Запись.ФизЛицо = Выборка.ФизЛицо;
	КонецЦикла;	
	//Запись в таблицу без замещения
	НаборЗаписей.Записать(Ложь);
КонецЦикла;

Далее меняем механизм расчета в ПроведениеРасчетовПереопределяемый:
1) ПолучитьСтруктуруНеобходимыхДанных – добавляем показатель СуммаПродаж.
2) ПолучитьДанныеДляРасчета:
	СуммаПродажТекст =
	"ВЫБРАТЬ РАЗЛИЧНЫЕ
	|	ОсновныеНачисленияОрганизаций.Ссылка,
	|	ОсновныеНачисленияОрганизаций.НомерСтроки
	|ПОМЕСТИТЬ ВТНачисленияУчитывающиеСуммуПродаж
	|ИЗ
	|	ПланВидовРасчета.ОсновныеНачисленияОрганизаций.Показатели КАК ОсновныеНачисленияОрганизаций
	|ГДЕ
	|	ОсновныеНачисленияОрганизаций.Показатель = ЗНАЧЕНИЕ(Справочник.ПоказателиСхемМотивации.СуммаПродаж)
	|;
	|
	|////////////////////////////////////////////////////////////////////////////////
	|ВЫБРАТЬ
	|	Основной.НомерСтроки КАК НомерСтроки,
	|	Основной.Организация КАК Организация,
	|	Основной.Сотрудник КАК Сотрудник,
	|	Основной.ВидРасчета КАК ВидРасчета
	|ПОМЕСТИТЬ ВТСотрудникиДляСуммыПродаж
	|ИЗ
	|	РегистрРасчета.ОсновныеНачисленияРаботниковОрганизаций КАК Основной
	|ГДЕ
	|	Основной.Регистратор = &парамРегистратор
	|	И Основной.Авторасчет
	|	И Основной.ВидРасчета В
	|			(ВЫБРАТЬ
	|				ВТНачисленияУчитывающиеСуммуПродаж.Ссылка
	|			ИЗ
	|				ВТНачисленияУчитывающиеСуммуПродаж КАК ВТНачисленияУчитывающиеСуммуПродаж)
	|;
	|
	|////////////////////////////////////////////////////////////////////////////////
	|ВЫБРАТЬ
	|	ВТСотрудникиДляСуммыПродаж.НомерСтроки КАК НомерСтроки,
	|	ПродажиРаботниковОбороты.СуммаОборот КАК СуммаПродаж,
	|	ВТНачисленияУчитывающиеСуммуПродаж.НомерСтроки КАК НомерПоказателяСуммаПродаж
	|ПОМЕСТИТЬ ВТСуммаПродаж
	|ИЗ
	|	ВТСотрудникиДляСуммыПродаж КАК ВТСотрудникиДляСуммыПродаж
	|		ЛЕВОЕ СОЕДИНЕНИЕ РегистрНакопления.ПродажиРаботников.Обороты(
	|				&парамПериодРегистрации,
	|				КОНЕЦПЕРИОДА(&парамПериодРегистрации, МЕСЯЦ),
	|				Период,
	|				(Организация, Сотрудник) В
	|					(ВЫБРАТЬ
	|						ВТСотрудникиДляСуммыПродаж.Организация,
	|						ВТСотрудникиДляСуммыПродаж.Сотрудник
	|					ИЗ
	|						ВТСотрудникиДляСуммыПродаж КАК ВТСотрудникиДляСуммыПродаж)) КАК ПродажиРаботниковОбороты
	|		ПО ВТСотрудникиДляСуммыПродаж.Организация = ПродажиРаботниковОбороты.Организация
	|			И ВТСотрудникиДляСуммыПродаж.Сотрудник = ПродажиРаботниковОбороты.Сотрудник
	|		ЛЕВОЕ СОЕДИНЕНИЕ ВТНачисленияУчитывающиеСуммуПродаж КАК ВТНачисленияУчитывающиеСуммуПродаж
	|		ПО (ВТНачисленияУчитывающиеСуммуПродаж.Ссылка = ВТСотрудникиДляСуммыПродаж.ВидРасчета)
	|
	|ИНДЕКСИРОВАТЬ ПО
	|	НомерСтроки";
А также:
Если НеобходимыеДанные.СуммаПродаж Тогда
	ТекстЗапроса = ТекстЗапроса + ", 
	|СотрудникиСуммаПродаж.НомерПоказателяСуммаПродаж Как НомерПоказателяСуммаПродаж";
КонецЕсли;
И еще:
Если НеобходимыеДанные.СуммаПродаж Тогда
	Запрос.Текст = СуммаПродажТекст;
	Запрос.Выполнить();
	СуммаПродажТекст = "ВТСуммаПродаж";
	ТекстЗапроса = ТекстЗапроса + "
	|ЛЕВОЕ СОЕДИНЕНИЕ ВТСуммаПродаж КАК СотрудникиСуммаПродаж
	|ПО Основной.НомерСтроки = СотрудникиСуммаПродаж.НомерСтроки";
КонецЕсли;
3) ПолучитьТекстПоказателя:
НеобходимыеДанные.СуммаПродаж или
А также:
?(НеобходимыеДанные.СуммаПродаж, "                 
|	КОГДА Показатели.Показатель1 = ЗНАЧЕНИЕ(Справочник.ПоказателиСхемМотивации.СуммаПродаж)
|		ТОГДА ЕстьNull(СотрудникиСуммаПродаж.СуммаПродаж,0)", "") +
4) РассчитатьЗаписиНабора, после «Если СтрокаТабличнойЧасти <> Неопределено тогда»:
Если ИсходныеДанные.НомерПоказателяСуммаПродаж<>null тогда
ИмяПоказателя = "Показатель"+ИсходныеДанные.НомерПоказателяСуммаПродаж;
	СтрокаТабличнойЧасти[ИмяПоказателя] = ИсходныеДанные[ИмяПоказателя];
	СтрокаДвижений[ИмяПоказателя] = ИсходныеДанные[ИмяПоказателя];
КонецЕсли;	
Цель: чтобы показатель писался, в том числе и при пересчете.

Отчет:
ВЫБРАТЬ
	ОсновныеНачисленияРаботниковОрганизаций.Сотрудник,
	ОсновныеНачисленияРаботниковОрганизаций.ПериодРегистрации,
	ОсновныеНачисленияРаботниковОрганизаций.Показатель1 КАК ПроцентВознаграждения,
	ОсновныеНачисленияРаботниковОрганизаций.Показатель2 КАК ОбъемПродаж,
	ВЫБОР
		КОГДА ОсновныеНачисленияРаботниковОрганизаций.Показатель2 * ОсновныеНачисленияРаботниковОрганизаций.Показатель1 > 5000
			ТОГДА ОсновныеНачисленияРаботниковОрганизаций.Результат
		ИНАЧЕ 0
	КОНЕЦ КАК Вознаграждение1,
	ВЫБОР
		КОГДА ОсновныеНачисленияРаботниковОрганизаций.Показатель2 * ОсновныеНачисленияРаботниковОрганизаций.Показатель1 > 5000
			ТОГДА 0
		ИНАЧЕ ОсновныеНачисленияРаботниковОрганизаций.Результат
	КОНЕЦ КАК НачисленоИсходяИзМинимальногоРазмераЗарплаты
ИЗ
	РегистрРасчета.ОсновныеНачисленияРаботниковОрганизаций КАК ОсновныеНачисленияРаботниковОрганизаций
ГДЕ
	ОсновныеНачисленияРаботниковОрганизаций.ВидРасчета = &ВидРасчета

[bookmark: _Toc349662379]Возможный код модуля формы (можно взять из ввода индивидуальных графиков к примеру) для случаев создания своих документов:
Код модуля формы:
Перем ДниНедели;

Процедура ПериодРегистрацииПриИзменении(Элемент)
РаботаСДиалогами.ДатаКакМесяцПодобратьДатуПоТексту(Элемент.Значение, ПериодРегистрации);
Элемент.Значение = РаботаСДиалогами.ДатаКакМесяцПредставление(ПериодРегистрации);
УстановитьВидимостьколонокТаблицыГрафикРаботы();
ОформитьЗаголовкиДнейМесяца();
КонецПроцедуры

Процедура ПериодРегистрацииНачалоВыбораИзСписка(Элемент, СтандартнаяОбработка)
РаботаСДиалогами.НачалоВыбораИзСпискаПредставленияПериодаРегистрации(Элемент, СтандартнаяОбработка, ПериодРегистрации, ЭтаФорма);
УстановитьВидимостьколонокТаблицыГрафикРаботы();
ОформитьЗаголовкиДнейМесяца();
КонецПроцедуры

Процедура ПериодРегистрацииОчистка(Элемент, СтандартнаяОбработка)
СтандартнаяОбработка = Ложь;
КонецПроцедуры

Процедура ПериодРегистрацииРегулирование(Элемент, Направление, СтандартнаяОбработка)
ПериодРегистрации = ДобавитьМесяц(ПериодРегистрации, Направление);
Элемент.Значение = РаботаСДиалогами.ДатаКакМесяцПредставление(ПериодРегистрации);
УстановитьВидимостьколонокТаблицыГрафикРаботы();
ОформитьЗаголовкиДнейМесяца();
КонецПроцедуры

Процедура ПериодРегистрацииАвтоПодборТекста(Элемент, Текст, ТекстАвтоПодбора, СтандартнаяОбработка)
РаботаСДиалогами.ДатаКакМесяцАвтоПодборТекста(Текст, ТекстАвтоПодбора, СтандартнаяОбработка);
КонецПроцедуры

Процедура ПериодРегистрацииОкончаниеВводаТекста(Элемент, Текст, Значение, СтандартнаяОбработка)
РаботаСДиалогами.ДатаКакМесяцОкончаниеВводаТекста(Текст, Значение, СтандартнаяОбработка);
КонецПроцедуры

Процедура ПриОткрытии()
Если ЭтоНовый() Тогда
	ЗаполнениеДокументовПереопределяемый.ЗаполнитьШапкуДокумента(ЭтотОбъект, глЗначениеПеременной("глТекущийПользователь"));
КонецЕсли;
МесяцСтрока = РаботаСДиалогами.ДатаКакМесяцПредставление(ПериодРегистрации);
УстановитьВидимостьколонокТаблицыГрафикРаботы();	
ОформитьЗаголовкиДнейМесяца();
КонецПроцедуры

Процедура ГрафикПосещенийСотрудникНачалоВыбора(Элемент, СтандартнаяОбработка)
ПроцедурыУправленияПерсоналомПереопределяемый.ОткрытьФормуВыбораСотрудникаОрганизации(Элемент, Ссылка, Истина, Дата, ОбщегоНазначения.ГоловнаяОрганизация(Организация), 3, СтандартнаяОбработка, Элемент.Значение);
КонецПроцедуры

Процедура ОформитьЗаголовкиДнейМесяца()
БледноКрасныйЦвет = Новый Цвет(255, 176, 176);
СерыйЦвет = Новый Цвет(234, 229, 216);
		
Запрос = Новый Запрос;
Запрос.УстановитьПараметр("НачалоМесяца",НачалоМесяца(ПериодРегистрации));
Запрос.УстановитьПараметр("КонецМесяца",КонецМесяца(ПериодРегистрации));
Запрос.Текст = 
"ВЫБРАТЬ
|	РегламентированныйПроизводственныйКалендарь.ВидДня,
|	ДЕНЬНЕДЕЛИ(РегламентированныйПроизводственныйКалендарь.ДатаКалендаря) КАК ДеньНедели
|ИЗ
|	РегистрСведений.РегламентированныйПроизводственныйКалендарь КАК РегламентированныйПроизводственныйКалендарь
|ГДЕ
|	РегламентированныйПроизводственныйКалендарь.ДатаКалендаря МЕЖДУ &НачалоМесяца И &КонецМесяца
|
|УПОРЯДОЧИТЬ ПО
|	РегламентированныйПроизводственныйКалендарь.ДатаКалендаря";
Результат = Запрос.Выполнить();
Если Результат.Пустой() Тогда
	ДеньНеделиМесяца = ДеньНедели(НачалоМесяца(ПериодРегистрации));
	Для ДеньМесяца = 1 по 10 Цикл
		Колонка = ЭлементыФормы.ГрафикПосещений.Колонки["День"+Строка(ДеньМесяца)];
		Колонка.ТекстШапки = Строка(ДеньМесяца) + Символы.ПС + ДниНедели[ДеньНеделиМесяца];
		Если ДеньНеделиМесяца = 6 Или ДеньНеделиМесяца = 7 Тогда
			Колонка.ЦветТекстаШапки = ЦветаСтиля.ЦветОсобогоТекста;
			Колонка.ШрифтШапки = ШрифтыСтиля.ШрифтВажнойНадписи;
		Иначе
			Колонка.ЦветТекстаШапки = Новый Цвет();	
			Колонка.ШрифтШапки = Новый Шрифт();
		КонецЕсли;
			ДеньНеделиМесяца = (ДеньНеделиМесяца%7)+1;
	КонецЦикла;
Иначе
	Данные = Результат.Выгрузить();
	ЦветСубботы	           = Новый Цвет(153, 51,   0); // Темно-красный
	ЦветВоскресенья 	   = ЦветаСтиля.ЦветОсобогоТекста; // Красный
	ЦветПредпразничногоДня = Новый Цвет(  0,  0, 186); // Темно-синий
	ЦветПраздничногоДня	   = Новый Цвет(255,  0, 255); // Фиолетовый
	Для ДеньМесяца = 1 по 10 Цикл
	Колонка = ЭлементыФормы.ГрафикПосещений.Колонки["День"+Строка(ДеньМесяца)];
Колонка.ТекстШапки = Строка(ДеньМесяца) + Символы.ПС + ДниНедели[Данные[ДеньМесяца - 1].ДеньНедели];
			
	ВидДня = Данные[ДеньМесяца - 1].ВидДня;
	Если ВидДня = Перечисления.ВидыДнейПроизводственногоКалендаря.Праздник Тогда
		Колонка.ЦветТекстаШапки = ЦветПраздничногоДня;
		Колонка.ШрифтШапки = ШрифтыСтиля.ШрифтВажнойНадписи;
	ИначеЕсли ВидДня = Перечисления.ВидыДнейПроизводственногоКалендаря.Воскресенье Тогда
		Колонка.ЦветТекстаШапки = ЦветВоскресенья;
		Колонка.ШрифтШапки = ШрифтыСтиля.ШрифтВажнойНадписи;
	ИначеЕсли ВидДня = Перечисления.ВидыДнейПроизводственногоКалендаря.Суббота Тогда
		Колонка.ЦветТекстаШапки = ЦветСубботы;
		Колонка.ШрифтШапки = ШрифтыСтиля.ШрифтВажнойНадписи;
	ИначеЕсли ВидДня = Перечисления.ВидыДнейПроизводственногоКалендаря.Предпраздничный Тогда
		Колонка.ЦветТекстаШапки = ЦветПредпразничногоДня;
		Колонка.ШрифтШапки = Новый Шрифт();
	Иначе
		Колонка.ЦветТекстаШапки = Новый Цвет();	
		Колонка.ШрифтШапки = Новый Шрифт();
	КонецЕсли;
	КонецЦикла;
КонецЕсли;
ЭлементыФормы.ГрафикПосещений.ВысотаШапки = 2;
КонецПроцедуры

Процедура УстановитьВидимостьколонокТаблицыГрафикРаботы()
ПоследнийДеньМесяца = День(КонецМесяца(ПериодРегистрации));
Для ДеньМесяца = 1 По 10 Цикл
	ВидимостьКолонок = ДеньМесяца <= ПоследнийДеньМесяца;
	ЭлементыФормы.ГрафикПосещений.Колонки["День"+Строка(ДеньМесяца)].Видимость = ВидимостьКолонок;
КонецЦикла;
КонецПроцедуры // УстановитьВидимостьколонокТаблицыГрафикРаботы()	
Процедура ПередЗаписью(Отказ, РежимЗаписи, РежимПроведения)
Для каждого стр из ГрафикПосещений Цикл
	итогоДней=0;
	Для сч=1 по 10 Цикл
		Если стр["День"+сч] тогда
			ИтогоДней = итогоДней + 1;
		КонецЕсли;
	КонецЦикла;
	стр.КоличествоДнейПосещения = ИтогоДней;
КонецЦикла;
КонецПроцедуры

ДлинаСуток = 86400;
ПерваяДата = НачалоДня(ОбщегоНазначения.ПолучитьРабочуюДату());
ПерваяДата = ПерваяДата - ДеньНедели(ПерваяДата) * ДлинаСуток;
ДниНедели = Новый Соответствие;
Для ДеньНедели = 1 По 7 Цикл
	ДниНедели.Вставить(ДеньНедели, НРег(Формат(ПерваяДата + ДеньНедели * ДлинаСуток,"ДФ=ддд")));
КонецЦикла;

